code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_11:墳墓の小迷宮
;-------------------------------------------------
@QST_TITLE_QUEST_11
RESULTS '= "淫魔の小迷宮"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_11
RESULTS '= "周期,探索,"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_11
RESULTS:0 '= "魔物・魔獣の中にはごく僅かだが人間と交流を持つ種が存在する"
RESULTS:1 '= "淫魔はその中でも代表的な種であり、辺境の一部地域には淫魔を子孫繁栄をもたらす存在として"
RESULTS:2 '= "信仰する地域さえあったという"
RESULTS:4 '= "淫魔は人間の精気を好んで喰らうことで知られているが、人間の性行動を娯楽的に消費することもあった"
RESULTS:5 '= "そこで一部の地域には淫魔に人間の性的痴態を娯楽として供し、その引き換えに"
RESULTS:6 '= "淫魔から様々な便宜を受け取る風習が存在したとされる"
RESULTS:8 '= "この奇妙な迷宮がその名残なのか、今となっては何もわからない"
RESULTS:9 '= "ただ一つ言えることは、身の清らかさを尊ぶならば、この迷宮に近寄るべきではないだろう"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_11
RESULTS:0 '= @"難易度：@C:{カラー_ピンク}@❤❤❤@/C@"
RESULTS:1 '= "報酬：-"
RESULTS:2 '= "パーティ人数:1人(女性のみ)　　　必要性技スキル：誘蛾の色香LV2以上"
RESULTS:3 '= "備考:定期性クエスト(4期毎)、退却可能　　　推奨平均戦闘値：45以上"
RESULTS:4 '= ""
RESULTS:5 '= @"@C:{カラー_選択不可}@「最近見つかった迷宮だ。俺も大したことは知らんよ」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「ただ確か……双子の淫魔を見かけたら力の限り逃げろとか、そんな話は耳にしたな」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_11(QUEST_ID)
#DIM QUEST_ID
;野盗討伐の攻略が必要
SIF QST_SV_QUEST_FLAG:12 != 1
    RETURN 0
;クエストのクールタイム。完了時に加算され、期終了時に減算される
SIF QST_COOLTIME:QUEST_ID
    RETURN 0
RETURN 1

;-------------------------------------------------
;クエスト固有の開始条件
;[このクエストを開始する]押下時に呼ばれる
;女性のみ・誘蛾の芳香LV2以上が必要
;-------------------------------------------------
@QST_CAN_START_QUEST_11(QUEST_ID)
#DIM QUEST_ID
#DIM DYNAMIC ERR_CNT
#DIM DYNAMIC CHARA_ID
LOCALS '= ""
;一人だけ
IF LIST_COUNT(_LIST_QST_PARTY) >= 2
    CALLF TAG_PRINT(@"@C:{カラー_注意}@一人のみ参加可能です@/C@", 1)
    WAIT
    RETURN 0
ENDIF
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF !IS_FEMALE(CHARA_ID)
        ERR_CNT ++
        LOCALS += @"<%ANAME(CHARA_ID)%>"
    ENDIF
    CALL SKILL_GET_LEVEL(CHARA_ID, スキル_大分類_性技, "SEDUCTION")
    IF RESULT < 2
        ERR_CNT ++
        LOCALS += @"<%ANAME(CHARA_ID)%>"
    ENDIF
NEXT
IF ERR_CNT
    CALLF TAG_PRINT(@"@C:{カラー_注意}@参加条件を満たしてないキャラが編成されています(女性かつ性技スキル:誘蛾の色香LV2以上が必要)@/C@", 1)
    CALLF TAG_PRINT(@"@C:{カラー_注意}@条件を満たしていないキャラ：%LOCALS%@/C@", 1)
    WAIT
    RETURN 0
ELSE
    RETURN 1
ENDIF

;-------------------------------------------------
;マップデータ定義
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_11(MAP_ID)
#DIM MAP_ID
QST_MAPNAME '= "淫魔の小迷宮"
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "淫魔の小迷宮"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,1,100,]"
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,100,]"
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,8,]"
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,9,100,100,100,]"
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,100,]"
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,7,6,4,]"
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,5,100,100,]"
_DIC_HO_MAPDATA:MAP_ID:8   '= "[EVENT_LIST:0,2,3,]"
_DIC_HO_MAPDATA:MAP_ID:9   '= "[EVENT_LIST:0,20,]"

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_11
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_11
#DIMS _LIST_COMMON_EVENT
#DIMS MAP_NAME

;コモンイベントをランダムに配置する
;宝箱
CALL QST_SET_RANDOM_EVENT(100, RAND(6, 9))
;戦闘
;CALL QST_SET_RANDOM_EVENT(105, RAND(5, 7))

FLAG:クエスト_中断を許可 = 1
FLAG:クエスト_刻限による終了無効 = 1

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_11_ABORT
QST_COOLTIME:QST_有効クエストID += 3

;-------------------------------------------------
; クエスト中全パーティメンバーが行動不能になった場合 @QST_TRAIN_EXIT より呼ばれる
;-------------------------------------------------
@QST_QUEST_11_PARTY_KNOCKED_OUT()
CALL QST_QUEST_11_DEFEAT_EVENT_2

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_11
PRINTFORML %QST_LEADER_NAME()%は依頼に従い、東方の辺境地で発見された古の遺跡を訪れた
PRINTFORML 山脈の麓に築かれたこの遺跡は、かつてはある部族の祭祀場だったという
PRINTFORMW その部族が滅び、幾ばくかの世紀が過ぎて人々の記憶からは忘れさられてしまったようだ
PRINTL
PRINTFORML 幾人かの冒険者が調査のために派遣されたが、帰ってきたものは皆虚ろな目をして何があったのか語ろうとしない
PRINTFORMW ……探索に臨むにあたり、十分に警戒すべきだろう
CALL FL_PLAYBGM("Ambient_dark_0.ogg")

;-------------------------------------------------
;イベントリストに登録するイベント情報
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 1
        CALL HO_PRINT_LOG("まぐわう男女の彫像が無造作に並べられている", 1)
    CASE 2
        CALL HO_PRINT_LOG("小箱や調度品、催事用の器具が並べられてる", 1)
    CASE 3
        CALL HO_PRINT_LOG("殺風景な、開けた空間がある。四隅には観覧席のようなものが設けられている", 1)
    CASE 4
        CALL HO_PRINT_LOG("宝飾品や磁器が積み上げられている", 1)
    CASE 6
        CALL HO_PRINT_LOG("書斎のようだ", 1)
        CALL HO_PRINT_LOG("部屋の中央に置かれた場違いなティーテーブルが異質な存在感を放っている", 1)
    CASE 8
        CALL HO_PRINT_LOG("細く長い回廊が奥まで続いている", 1)
    CASEELSE
ENDSELECT
;-------------------------------------------------
;EV1
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_1
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_1
#DIM EVENT_ID = 2

PRINTFORML 遺跡の中は暗く、狭い
PRINTFORMW 携えた光源の頼りない光芒が深淵へと続く石造りの壁を照らし出す
PRINTL
PRINTFORML 壁面は悠久の年月を経て風化しところどころ掛けたり、崩れたりしているものの、
PRINTFORMW ほぼ隙間なく組まれておりこの遺跡が高度な建築技術を持って造られたことを伺わせる
PRINTL
PRINTFORMW %QST_LEADER_NAME()%はあちこちに無数の石像が据え置かれていることに気がついた
PRINTL
PRINTFORML そのいずれも性交や姦淫、乱交をモチーフにしており、%QST_LEADER_NAME()%は面を食らったが
PRINTFORMW しかし造形の精巧さに掛けては息を飲むものであり、吐息や嬌声さえ聞こえんばかりの生々しさがあった
PRINTL
PRINTFORML 今のところ生物の気配は感じられない。しかし奇妙な、誰かから見られているような落ち着かない感触があった
PRINTFORMW 周囲の彫像のせいだろうか……？　いずれにせよ注意して進んだ方がよさそうだ

;-------------------------------------------------
;EV2
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_2
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_SLOT_NAME = "リビングロープの罠"
#DIMS EVENT_TYPE = "ONCE"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_2
#DIM CONST EVENT_ID = 2
#DIM EVENT_FLAG
#DIM CHARA_ID

CHARA_ID = QST_RANDCHARA()

CALL COLOR_PRINTW(@"―――― 部屋に入るなり、縄が意志を持ったように%ANAME(CHARA_ID)%に一斉に絡みついた！", カラー_注意)
CALL PRINT_TITLE("リビングロープの罠", "エロトラップ")
PRINTW
CALL QST_ROLL_MULTI("戦闘,身体,", "素早くナイフで断ち切る(戦闘値),全力で振りほどく(身体),","500,550,", 7, ,"なすがまま", CHARA_ID)
IF RESULT == 1
    PRINTFORMW %ANAME(CHARA_ID)%は生ける縄の襲撃を辛うじて避けることができた……
    CALL QST_DEACTIVATE_EVENT(3)
ELSE
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID", TOSTR(CHARA_ID)
    PRINTFORMW %ANAME(CHARA_ID)%は生ける縄に捉えられ天井に吊り上げられてしまった……
    PRINTL
    PRINTFORML もがき、逃れようとする%ANAME(CHARA_ID)%に部屋の奥までピンと張り詰めた縄が一本渡される
    PRINTFORMW 縄は逃れようとも逃れようとも%ANAME(CHARA_ID)%が無理やり縄に跨る格好となるよう強要してくる
    PRINTL
    PRINTFORML %ANAME(CHARA_ID)%は一本の縄の先、部屋の奥にスイッチがあるのを認めた
    PRINTFORMW 実にわざとらしいが……今はあれが解除用のスイッチと祈る他ないだろう……
    PRINTL
    PRINTFORMW 渡し縄にはねっとりとした潤滑油が塗られており、所々に結び目が設けられている
    PRINTFORMW この罠は明らかに侵入者を捕えるのではなく、捉えた者を苦悶させ痴態を晒させることを意図しているようだ……
    CALL QST_ACTIVATE_EVENT(3)
    FLAG:移動制限フラグ = 2
    QST_イベント選択制限 = 3
    CALL QST_SET_EVENT_FLAG(3, 7)
    CALL HO_PRINT_LOG(@"罠に捉えられてしまった……。今はこの股縄を進む他ない……", 1)
ENDIF

;-------------------------------------------------
;EV3 股縄渡り
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_3
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "股縄を進む"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_3
#DIM CONST EVENT_ID = 3
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(3)
EVENT_FLAG = RESULT

CHARA_ID = TOINT(MAP_GET(@"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID"))
IF EVENT_FLAG > 1
    IF EVENT_FLAG != 3
        PRINTFORML %ANAME(CHARA_ID)%は股縄の擦れる感覚に耐えながら、一歩ずつ前へ進む
        PRINTFORMW たっぷりと潤滑油が塗られた縄に設けられた結び目が陰部の敏感な箇所を刺激する……
        ;ソース処理
        SOURCE:CHARA_ID:反感 = 100
        SOURCE:CHARA_ID:不安 = 10
        SOURCE:CHARA_ID:逸脱 = 50
        SOURCE:CHARA_ID:快Ｃ += SENSE_HOUSHI(CHARA_ID, CHARA_ID, 150)
    ELSE
        CALL COLOR_PRINTW(@"ようやく中央まで進んだところで、%ANAME(CHARA_ID)%の跨る縄が激しく前後左右に振動する！", カラー_注意)
        CALL COLOR_PRINTW(@"息絶え絶えになる%ANAME(CHARA_ID)%の股間を荒縄が激しく刺激する……", カラー_ピンク)
        ;ソース処理
        SOURCE:CHARA_ID:反感 = 150
        SOURCE:CHARA_ID:不安 = 30
        SOURCE:CHARA_ID:逸脱 = 80
        SOURCE:CHARA_ID:快Ｃ += SENSE_HOUSHI(CHARA_ID, CHARA_ID, 450)
    ENDIF
    CALL QST_EXEC_DUMMYCOM()
    EVENT_FLAG --
    CALL QST_SET_EVENT_FLAG(3, EVENT_FLAG)
    CALL HO_CREAR_LOG()
    CALL HO_PRINT_LOG(@"渡り切るまであと@C:{カラー_黄}@{EVENT_FLAG}歩@/C@は必要だろう……", 1)
ELSE
    PRINTFORMW %ANAME(CHARA_ID)%は性感への執拗な刺激になんとか耐えながら足でスイッチを蹴る
    PRINTFORMW スイッチを解除すると、縄は力を失いあっけなく解けた
    PRINTL
    PRINTFORMW 全く単なる嫌がらせでしかない罠に悪態をつきつつ、%ANAME(CHARA_ID)%は先へと進む……
    CALL HO_CREAR_LOG()
    CALL QST_DEACTIVATE_EVENT(EVENT_ID)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
ENDIF

;-------------------------------------------------
;EV4
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_4
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "テーブルの上の古書物"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_4
#DIM CONST EVENT_ID = 4
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(EVENT_ID)
IF !RESULT
    CHARA_ID = QST_RANDCHARA()
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID", TOSTR(CHARA_ID)
    PRINTFORML %ANAME(CHARA_ID)%はティーテーブルに置かれた書籍を手に取った
    PRINTFORMW 表紙は非常に凝った意匠の装飾が施されている。どうやら魔術書の類のようだ
    PRINTFORMW ……そして、この遺跡の調度品の例に漏れず、淫らな意匠が施されている
    PRINTL
    PRINTFORMW %ANAME(CHARA_ID)%はその本を調べてみてもよいし、無視してもよい
    CALL ASK_YN("手にとって調べてみる", "無視する")
    PRINTL
    IF !RESULT
        PRINTFORMW %ANAME(CHARA_ID)%は本を手に取り、ページをめくる……
        PRINTL
        PRINTFORMW 本の中は……びっしりと性行為に耽る男女の絵図で埋め尽くされている
        PRINTFORMW 非常に緻密な、まるで空間をそのまま切り出したかのような精緻な点描で描かれており
        PRINTFORMW あまりにも生々しい描写に息を飲む他ない
        PRINTL
        PRINTFORMW %ANAME(CHARA_ID)%は吸い込まれるようにページをめくる……
        PRINTFORMW 男女のまぐわいから始まり、輪姦、乱交、獣姦……おおよそありとあらゆる性交渉の場面が描かれているように思われた
        PRINTFORMW 無心にページをめくる、めくる、めくる……
        PRINTFORMW
        PRINTFORMW めくる、めくる、めくる……
        PRINTFORML
        CALL COLOR_PRINTW(@"%ANAME(CHARA_ID)%は自分の意志と関わらず、ページをめくる手が止まらないことに気がついた……！", カラー_注意)
        CALL PRINT_TITLE("淫魔の催眠魔術書", "エロトラップ")
        CALL QST_ROLL_MULTI("魔術,", "意志の力で抵抗する(魔術),","500,", , ,"なすがまま", CHARA_ID)
        IF RESULT == 1
            PRINTFORMW %ANAME(CHARA_ID)%はなんとか書籍を握る手を払い除け、本の魔力から脱することに成功した……
        ELSE
            CALL COLOR_PRINTW(@"―――― %ANAME(CHARA_ID)%は本の魔力に捉えられてしまった……！", カラー_注意)
            CALL COLOR_PRINTW(@"下腹部が焼けるように熱を帯びる……。手が自分の意志とは関係なく自身を慰めようとする……", カラー_ピンク)
            FLAG:移動制限フラグ = 2
            QST_イベント選択制限 = EVENT_ID
            CALL QST_SET_EVENT_SLOT_NAME(, @"@C:{カラー_ピンク}@❤魔術書のページをめくる❤@/C@")
            CALL QST_SET_EVENT_FLAG(EVENT_ID, 1)
            TCVAR:CHARA_ID:催眠中 = 1
            TCVAR:CHARA_ID:催眠実行者 = -1
            CALL CLEAR_ACTOR
            CALL ADD_MPLY(CHARA_ID)
            CALL ADD_MTAR(CHARA_ID)
            IF HAS_PENIS(CHARA_ID)
                CALL QST_EXEC_DUMMYCOM(T_N("Ｐ自慰"))
            ELSE
                CALL QST_EXEC_DUMMYCOM(T_N("Ｃ自慰"))
            ENDIF
            CALL HO_PRINT_LOG(@"ページをめくる手が止まらない……。自分を慰める手が止まらない……", 1)
            MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_11_EVENT_4_RELEASE_EVENT"

        ENDIF
    ELSE
        PRINTFORMW %ANAME(CHARA_ID)%は本を元の場所に戻した……
    ENDIF
ELSEIF RESULT == 1
    CHARA_ID = TOINT(MAP_GET(@"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID"))
    PRINTFORMW %ANAME(CHARA_ID)%は食い入るようにページを捲りながら、あられもない痴態をさらし、自身を慰めている……
    CALL CLEAR_ACTOR
    CALL ADD_MPLY(CHARA_ID)
    CALL ADD_MTAR(CHARA_ID)
    IF HAS_PENIS(CHARA_ID)
        CALL QST_EXEC_DUMMYCOM(T_N("Ｐ自慰"))
    ELSE
        CALL QST_EXEC_DUMMYCOM(T_N("Ｃ自慰"))
    ENDIF
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_11_EVENT_4_RELEASE_EVENT"
ELSEIF RESULT == 2

ENDIF

;コマンド実行後処理 @EVENTCOMEND で呼び出される。コマンド実行後に催眠が解除されたかを見る
;催眠解除でイベント終了
@QST_QUEST_11_EVENT_4_RELEASE_EVENT
#DIM CHARA_ID

CHARA_ID = TOINT(MAP_GET(@"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID"))

;長すぎてもダレるので10ターンで強制解除
IF TCVAR:CHARA_ID:催眠経過ターン数 == 10 || !IS_PLAYABLE(CHARA_ID)
    CALL COLOR_PRINTL(@"%ANAME(CHARA_ID)%の催眠が解けた！", カラー_注意)
    TCVAR:CHARA_ID:催眠中 = -1
    TCVAR:CHARA_ID:催眠経過ターン数 = 0
ENDIF

IF TCVAR:CHARA_ID:催眠中 == -1
    PRINTL
    CALL COLOR_PRINTW(@"―――― %ANAME(CHARA_ID)%はハッと我に返り、握りしめていた魔術書を投げ捨てた……！", カラー_注意)
    CALL QST_CHANGE_ITEM("淫堕点,15,")
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    CALL RELEASE_ALL_EQUIP(CHARA_ID)
    CALL QST_SET_EVENT_FLAG(4, 0)
    CALL QST_DEACTIVATE_THIS_EVENT
    CALL HO_CREAR_LOG()
ELSE
    PRINTL
    PRINTFORML %ANAME(CHARA_ID)%は本の魔力に取り憑かれてしまっている！
    PRINTFORMW 今は本を読み進めるより他はない……
ENDIF

;-------------------------------------------------
;EV5
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_5
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "彫像の手に握られた古書物"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_5
#DIM CONST EVENT_ID = 5
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(EVENT_ID)
IF !RESULT
    CHARA_ID = QST_RANDCHARA()
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID", TOSTR(CHARA_ID)
    PRINTFORMW %ANAME(CHARA_ID)%は、一糸まとわぬ姿で右手を下腹部に這わせ、恍惚とした表情を浮かべる女の像の前に立った
    PRINTFORMW 彫像は両性具有を模して掘られている。女の右手は、不釣り合いに巨大に励起した男性器にあてがわれている……
    PRINTL
    PRINTFORMW 彫像は左手に非常に凝った意匠の書籍を握りしめている。どうやら魔術書の類のようだ
    PRINTL
    PRINTFORMW %ANAME(CHARA_ID)%はその本を調べてみてもよいし、無視してもよい
    CALL ASK_YN("手にとって調べてみる", "無視する")
    PRINTL
    IF !RESULT
        PRINTFORMW %ANAME(CHARA_ID)%は本を手に取り、ページをめくる……
        PRINTL
        PRINTFORMW 本の中は……びっしりと性行為に耽る女同士の絵図で埋め尽くされている
        PRINTFORMW 奇妙なことにどの女も両性具有として描かれている……
        PRINTFORMW 非常に緻密な、まるで空間をそのまま切り出したかのような精緻な点描で描かれており
        PRINTFORMW あまりにも生々しい描写に息を飲む他ない
        PRINTL
        PRINTFORMW %ANAME(CHARA_ID)%は吸い込まれるようにページをめくる……
        PRINTFORMW 女と女がひたすらにまぐわい、犯し、快楽に耽る様が描かれている……
        PRINTFORMW 無心にページをめくる、めくる、めくる……
        PRINTFORMW
        PRINTFORMW めくる、めくる、めくる……
        PRINTFORML
        CALL COLOR_PRINTW(@"%ANAME(CHARA_ID)%は自分の意志と関わらず、ページをめくる手が止まらないことに気がついた……！", カラー_注意)
        CALL PRINT_TITLE("淫魔のふたなり催眠魔術書", "エロトラップ")
        CALL QST_ROLL_MULTI("魔術,", "意志の力で抵抗する(魔術),","500,", , ,"なすがまま", CHARA_ID)
        IF RESULT == 1
            PRINTFORMW %ANAME(CHARA_ID)%はなんとか書籍を握る手を払い除け、本の魔力から脱することに成功した……
        ELSE
            CALL COLOR_PRINTW(@"―――― %ANAME(CHARA_ID)%は本の魔力に捉えられてしまった……！", カラー_注意)
            CALL COLOR_PRINTW(@"下腹部が焼けるように熱を帯びる……。手が自分の意志とは関係なく自身を慰めようとする……", カラー_ピンク)
            IF !HAS_PENIS(CHARA_ID)
                PRINTL
                CALL COLOR_PRINTW(@"……更に奇妙なことに、自身の股間からあり得ないモノがぶら下がっているのを目の当たりにした", カラー_白)
                CALL COLOR_PRINTW(@"固く励起した男根だ……。先端から絶え間なくカウパーを垂れ流し、焼き切れるような劣情が下腹部からせり上がってくる……", カラー_白)
                PRINTL
                CALL COLOR_PRINTW(@"―――― %ANAME(CHARA_ID)%は一時的にふたなりにされてしまった……！", カラー_ピンク)
                MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "ふたなり化", TOSTR(CHARA_ID)
                CALL TO_FUTANARI(CHARA_ID)
                CALL SET_PENISSIZE(CHARA_ID, 1)
                CFLAG:CHARA_ID:一時的ふたなり = 1
            ENDIF
            FLAG:移動制限フラグ = 2
            QST_イベント選択制限 = EVENT_ID
            CALL QST_SET_EVENT_SLOT_NAME(, @"@C:{カラー_催眠}@❤魔術書のページをめくる❤@/C@")
            CALL QST_SET_EVENT_FLAG(EVENT_ID, 1)
            TCVAR:CHARA_ID:催眠中 = 1
            TCVAR:CHARA_ID:催眠実行者 = -1
            CALL CLEAR_ACTOR
            CALL ADD_MPLY(CHARA_ID)
            CALL ADD_MTAR(CHARA_ID)
            CALL QST_EXEC_DUMMYCOM(T_N("Ｐ自慰"))
            CALL HO_PRINT_LOG(@"ページをめくる手が止まらない……。自分を慰める手が止まらない……", 1)
            MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_11_EVENT_4_RELEASE_EVENT"

        ENDIF
    ELSE
        PRINTFORMW %ANAME(CHARA_ID)%は本をそのままにし、その場を離れた……
    ENDIF
ELSEIF RESULT == 1
    CHARA_ID = TOINT(MAP_GET(@"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID"))
    PRINTFORMW %ANAME(CHARA_ID)%は食い入るようにページを捲りながら、あられもない痴態をさらし、自身を慰めている……
    CALL CLEAR_ACTOR
    CALL ADD_MPLY(CHARA_ID)
    CALL ADD_MTAR(CHARA_ID)
    IF HAS_PENIS(CHARA_ID)
        CALL QST_EXEC_DUMMYCOM(T_N("Ｐ自慰"))
    ELSE
        CALL QST_EXEC_DUMMYCOM(T_N("Ｃ自慰"))
    ENDIF
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_11_EVENT_5_RELEASE_EVENT"
ELSEIF RESULT == 2

ENDIF

;コマンド実行後処理 @EVENTCOMEND で呼び出される。コマンド実行後に催眠が解除されたかを見る
;催眠解除でイベント終了
@QST_QUEST_11_EVENT_5_RELEASE_EVENT
#DIM CHARA_ID

CHARA_ID = TOINT(MAP_GET(@"QUEST_{QST_有効クエストID}_MAP", "CHARA_ID"))
;長すぎてもダレるので10ターンで強制解除
IF TCVAR:CHARA_ID:催眠経過ターン数 == 10 || !IS_PLAYABLE(CHARA_ID)
    CALL COLOR_PRINTL(@"%ANAME(CHARA_ID)%の催眠が解けた！", カラー_注意)
    TCVAR:CHARA_ID:催眠中 = -1
    TCVAR:CHARA_ID:催眠経過ターン数 = 0
ENDIF

IF TCVAR:CHARA_ID:催眠中 == -1
    PRINTL
    CALL COLOR_PRINTW(@"―――― %ANAME(CHARA_ID)%はハッと我に返り、握りしめていた魔術書を投げ捨てた……！", カラー_注意)
    FLAG:移動制限フラグ = 0
    QST_イベント選択制限 = -1
    CALL RELEASE_ALL_EQUIP(CHARA_ID)
    ;ふたなり化されている場合解除
    IF MAP_HAS(@"QUEST_{QST_有効クエストID}_MAP", "ふたなり化")
        CALL TO_FUTANARI(CHARA_ID)
        MAP_REMOVE @"QUEST_{QST_有効クエストID}_MAP", "ふたなり化"
        CFLAG:CHARA_ID:一時的ふたなり = 0
        CALL COLOR_PRINTW(@"%ANAME(CHARA_ID)%の股間に生えていたモノはいつの間にか消えていた……", カラー_注意)
        CALL QST_CHANGE_ITEM("淫堕点,20,")
    ENDIF
    CALL QST_SET_EVENT_FLAG(5, 0)
    CALL QST_DEACTIVATE_THIS_EVENT
    CALL HO_CREAR_LOG()
ELSE
    PRINTL
    PRINTFORML %ANAME(CHARA_ID)%は本の魔力に取り憑かれてしまっている！
    PRINTFORMW 今は本を読み進めるより他はない……
ENDIF

;-------------------------------------------------
;EV6
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_6
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "卓上の水差し"
#DIMS EVENT_TYPE = "ONCE"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_6
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %ANAME(QST_RANDCHARA())%はティーテーブル上の水差しを覗き込んだ
PRINTFORMW 中には並々と水が注がれている
PRINTL
PRINTFORML なんとも怪しい……しかし、もしも手ひどく消耗しているなら背に腹は変えられない
PRINTFORML 機会は一度きりだが、喉を潤し身体を休めることができる……かもしれない
PRINTL
CALL ASK_YN("喉を潤す", "やめておく")
IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%は喉の乾きを癒やし、小休憩を取った
    ;パーティ全員の体力を回復
    FOR LOCAL, 0, QST_PARTYNUM
        TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
        CALL QST_CHANGE_HP(TARGET_ID, 1500)
        TCVAR:TARGET_ID:媚薬 += 1
    NEXT
    PRINTL
    PRINTFORMW ……
    PRINTFORMW ……%QST_LEADER_NAME()%は身体に奇妙な火照りを覚えた
    PRINTFORMW 鼓動が早くなる……。息が荒くなる……
    PRINTL
    PRINTFORMW 下腹部が火照り、情欲が頭を満たしていく……
    PRINTL
    CALL COLOR_PRINTW(@"―――― %QST_LEADER_NAME()%は媚薬により発情してしまった……！", カラー_ピンク)
    CALL QST_DEACTIVATE_THIS_EVENT()
ELSE
ENDIF

;-------------------------------------------------
;EV7
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_7
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_SLOT_NAME = ""

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_7
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID

PRINTFORML %QST_LEADER_NAME()%が足を踏み入れた部屋は、どうやら書斎のようだ
PRINTFORMW 四方の壁は朽ちた本がぎっしりと詰まった書棚で囲まれている
PRINTL
PRINTFORML 部屋の中心には奇妙なティーテーブルが置かれていた
PRINTFORMW 今日のノルディアで好んで使われる、流線型の装飾が施されたティーテーブルで、異質で場違いな存在感を放っている
PRINTL
PRINTFORML テーブルの上には二組のティーセットと水差しが置かれている
PRINTFORMW あたかもつい今しがたまで誰かがここで茶会を催していたかのようだ……

;-------------------------------------------------
;EV8 オートマタ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_8
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_SLOT_NAME = ""
#DIMS EVENT_TYPE = "ONCE"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_8
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID

PRINTFORML %QST_LEADER_NAME()%が足を踏み入れた部屋はこれまでとうって変わって、殺風景で広々としている
PRINTFORMW 部屋の四隅には観覧用と思しき席が設けられており、中央には大人の背丈ほどの鉄の人形が鎮座しているが……
PRINTL
PRINTFORMW ――――！　%QST_LEADER_NAME()%は人形の正体を認めるや即座に武器を握りしめた。これは冒険者の脅威として名高い自動人形の類だ
PRINTFORML 魔術で動作する人形で、単純な動作を繰り返す原始的なものからヒトを模した複雑なものまで多種多様だが、
PRINTFORMW 一度敵対したならば、この命も感情も持たない存在は壊れるまで戦うことを止めないだろう
PRINTL
PRINTFORML 案の定自動人形はギシギシと軋む音を立てて動作を始める
PRINTFORMW 禍々しい仮面に覆われた顔が%QST_LEADER_NAME()%に向けられる。脚として備えられた鋼の車輪が回転を始める……
PRINTFORMW 戦闘用か、あるいは拷問用か。老朽化しているものの、深刻な脅威であることは間違いない
PRINTL
PRINTFORML 眼前の驚異を排除しなければならない
PRINTFORMW 真正面から戦うことも出来るが、仲間のいないこの状況では困難な戦いとなるだろう
PRINTFORMW あるいは……何か別の策を講じることもできるかもしれないが……
PRINTL
CALL PRINT_TITLE("古代のオートマタ","トラップ")
PRINTFORML なにか打開策はないだろうか……？　腕に覚えがあるならば実力で脅威を排してもよい
PRINTL
CALL QST_ROLL_MULTI("観察,器用","部屋を観察してみる(観察),露出した機構部めがけて投石する(器用),", "80,100,", 7, 1, "戦闘を挑む")
IF RESULT == 1
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW %ANAME(RESULT:1)%は部屋の隅、観覧席にレバーがあることを見出した！
            PRINTL
            PRINTFORML 恐らく人形の主が高みから動作を制御するために設けたものだろう
            PRINTFORMW 幸いにして今は人形と自分の他は誰もいない。%ANAME(RESULT:1)%は脱兎の如くレバーに飛び寄り、体重を乗せて引き倒す
            PRINTW
            PRINTFORMW 動力を断たれた自動人形のアームがだらりと下がり、動作を止める。どうやら停止させることに成功したようだ！
            CALL QST_CHANGE_ITEM("名誉点,20,")
            RETURN
        CASE 1
            PRINTFORMW %ANAME(RESULT:1)%は落ちていた小石を拾うと、自動人形の欠けた装甲板から露出する機構めがけて投げつけた！
            PRINTFORML 複雑怪奇な歯車部品の間に石が挟まり込み、メキメキと嫌な音を立てる
            PRINTFORMW 程なくして歯車の回転が止まり、自動人形は糸が切れた操り人形のようにうなだれた
            PRINTW
            PRINTFORMW ……幸運にも自動人形を停止させることに成功したようだ！
            CALL QST_CHANGE_ITEM("名誉点,20,")
            RETURN
    ENDSELECT
ELSEIF !RESULT
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW ……%ANAME(RESULT:1)%は有効な手口を何も見出すことができなかった……
        CASE 1
            PRINTFORMW ……%ANAME(RESULT:1)%はやけくそ気味に石を投げつけてみたものの、石は装甲板に阻まれ、むなしく音を立てて落ちた……
    ENDSELECT
    PRINTFORMW もはや小細工を弄す余裕はない……！
    PRINTL
    PRINTFORML %QST_LEADER_NAME()%は武器を固く握りしめ、襲い来る鋼の敵に対峙した
    PRINTFORMW 侵入者の存在を認めた自動人形はアームを自在に振るい、%QST_LEADER_NAME()%めがけて鞭を振り下ろさんとしている……！
ELSEIF RESULT == -1
    PRINTFORML %QST_LEADER_NAME()%は武器を固く握りしめ、意思を持たぬ鋼の人形に襲いかかった
    PRINTFORMW 侵入者の存在を認めた自動人形はアームを自在に振るい、%QST_LEADER_NAME()%めがけて鞭を振り下ろさんとしている……！
ENDIF

;戦闘イベント
CALL CC_INIT()
CALL CC_ADDMOB("DEFENDER", "オートマタ・トーメンタ", 3000, 3000, 55, 0, 1)
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
NEXT
CALL CLOSE_COMBAT()
IF RESULT == 1
    CALL CC_ADD_COMBAT_EXP(7)
    CALL CC_APPLY_CHARA_HP()
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%は実力を以て立ちはだかる脅威を粉砕した！", カラー_注意)
    CALL QST_CHANGE_ITEM("名誉点,15,")
ELSE
    CALL CC_APPLY_CHARA_HP()
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%の武器を自動人形のアームが弾き飛ばした！", カラー_注意)
    PRINTFORML ……万策尽きてたじろぐ%QST_LEADER_NAME()%に向けて自動人形が縄を放つ
    PRINTFORMW %QST_LEADER_NAME()%は瞬く間に捕縛され、自由を失ってしまった……
    PRINTL
    PRINTFORMW 床に伏せる%QST_LEADER_NAME()%めがけ、続けて鞭が振り下ろされる
    PRINTFORMW どうやらこの魂なき拷問官に慈悲は期待できそうになさそうだ……
    PRINTL
    CALL PRINT_TITLE("自動人形の仕置き","エロイベント")
    WAIT
    CALL QST_汎用イベント調教_機械姦ＳＭ(12)
    CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(14)
ENDIF

;-------------------------------------------------
;EV9 インキュバスの扉
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_9
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_SLOT_NAME = "インキュバスの扉"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_9
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID
#DIM 射精回数
CALL QST_GET_EVENT_FLAG(9)
射精回数 = RESULT

CALL PRINT_TITLE("雄々しく勃ける彫像の扉", "エロイベント")
CALL QST_GET_EVENT_FLAG(10)
IF !RESULT
    PRINTFORML %QST_LEADER_NAME()%の前に分厚い石造りの扉が立ちはだかる
    PRINTFORML 扉にはインキュバスを模したと思われる、精巧な彫像が穿たれている
    PRINTFORMW 今にも動き出さんばかりの精巧さだ
    PRINTL
    PRINTFORMW ……そしてとりわけ目を引くのが、像の股間からいきり立った巨大な雄棒だ…
    PRINTFORMW 眼前にそそり立つそれは、血管に至るまで緻密なディテールで彫り込まれている
    PRINTL
    PRINTFORML %QST_LEADER_NAME()%はこの扉にノブや錠の類が一切ないことに気がついた
    PRINTFORMW よく扉を見回してみると、像の頭上に文字が彫り込まれていることを認めた
    PRINTL
ELSE
    PRINTFORML %QST_LEADER_NAME()%は再び淫魔の像の前に立った
ENDIF
PRINTFORMW 扉には『汝の舌技を示せ』とある……
PRINTL
CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, 0))
CALL CHECK_COM_KNOWLEDGE(CHARA_ID, T_N("フェラチオ"))
IF !RESULT
    PRINTFORMW しかし性知識に疎い%ANAME(CHARA_ID)%には何のことか理解できなかった……
    CALL HO_MOVE_MASTER(3)
    RETURN
ENDIF

CALL ASK_YN("像の淫棒に唇をつけてみる", "放っておく")
IF RESULT
    PRINTFORMW 馬鹿げている……。放っておこう……
    CALL HO_MOVE_MASTER(3)
    RETURN
ENDIF

PRINTFORMW %ANAME(CHARA_ID)%が恐る恐る彫像の陰部に唇を当てると、像のペニスが仄かに明滅した……
PRINTL
PRINTFORMW この仕掛の主が何を欲しているのか理解した%ANAME(CHARA_ID)%は呆れつつも、所詮は作り物だと言い聞かせ"奉仕"を始めた……
CALL COLOR_PRINTL("TIPS:もしかしたらキャラクターは奉仕を拒むかもしれない", カラー_黄)
CALL COLOR_PRINTW("その場合は{特殊}→コマンド:なにもしない　でやり過ごさせよう", カラー_黄)

VARSET COM_FORBIDDEN:CHARA_ID:0, 1
;竿役生成
FOR LOCAL, 0, 1
    CALL QST_QUEST11_CREATE_MOB(0, "インキュバスの像")
    VARSET COM_FORBIDDEN:CHARA_ID:0, 1
    CHARA_ID = RESULT
    CFLAG:CHARA_ID:調教参加フラグ = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "MOB_ID", @"{CHARA_ID}"
NEXT

FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    VARSET COM_FORBIDDEN:CHARA_ID:0, 1
    COM_FORBIDDEN:CHARA_ID:T_N("フェラチオ") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("亀頭責めフェラ") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("なにもしない") = 0
    CFLAG:CHARA_ID:調教参加フラグ = 1
NEXT
CALL RESET_ACTOR()
FLAG:ウフフフラグ = 1
QST_調教ターン数 = 8
TFLAG:調教ターン数 = QST_調教ターン数
FLAG:調教モード = 調教_閨
TFLAG:交流中断不可 = 1
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE", "QST_QUEST_11_EVENT_9_COMENDEVENT"
CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(10)

;像の射精回数をカウント
@QST_QUEST_11_EVENT_9_COMENDEVENT
#DIM MOB_ID
#DIM 射精回数
MOB_ID = TOINT(MAP_GET(@"QUEST_{QST_有効クエストID}_MAP", "MOB_ID"))
IF NOWEX:MOB_ID:射精 > 0
    CALL QST_GET_EVENT_FLAG(9)
    射精回数 = RESULT + 1
    CALL QST_SET_EVENT_FLAG(9, 射精回数)
ENDIF
CALL HO_CREAR_LOG()
CALL HO_PRINT_LOG(@"インキュバスの像を射精させた回数：@C:{カラー_ピンク}@{射精回数}回@/C@", 1)

;-------------------------------------------------
;EV10 インキュバスの扉-調教後
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_10
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_10
#DIM 射精回数
#DIM CHARA_ID

MAP_REMOVE @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE"
CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, 0))
CALL QST_GET_EVENT_FLAG(9)
射精回数 = RESULT
IF 射精回数 < 1
    PRINTFORMW …………
    PRINTFORMW 像は微動だにしない……
ELSEIF 射精回数  < 3
    PRINTFORMW 像のペニスから白く粘性のある液体が程走り%ANAME(CHARA_ID)%を汚していく……
    PRINTL
    PRINTFORMW その青臭さ、生温かくねっとりとした感触は本物の精液そのものであった
    PRINTL
    PRINTFORMW 像の陰茎は怪しくピンク色に明滅している。だが、まだ不十分なようだ……
ELSE
    PRINTFORMW 像のペニスから白く粘性のある液体が程走り、%ANAME(CHARA_ID)%を汚していく……
    PRINTL
    PRINTFORMW その青臭さ、生温かくねっとりとした感触は精液そのものであった
    PRINTL
    PRINTFORML %ANAME(CHARA_ID)%の舌技で像のペニスがびくりと跳ね、絞り出すように白くベタつく液を吐き出した
    PRINTFORMW そして満足したかのように鎌首を下げると、同時に振動が起こる……
    PRINTL
    PRINTFORMW 扉がゆっくりと開く。道は開かれたようだ
    PRINTL
    PRINTFORMW はたしてこの先にあるものが%ANAME(CHARA_ID)%を汚辱しただけの価値があるものなのかは、今はわからない……
    CALL QST_CHANGE_ITEM("淫堕点,20,")
    CALL QST_DEACTIVATE_EVENT(9)
    CALL HO_CREAR_LOG()
    RETURN

ENDIF
CALL QST_SET_EVENT_FLAG(10, 1)
CALL HO_CREAR_LOG()
CALL HO_MOVE_MASTER(3)

;-------------------------------------------------
;EV14 トーメンタ-調教後
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_14
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_14
#DIM CHARA_ID

CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, 0))
PRINTFORML %QST_LEADER_NAME()%に激しい責め苦を加え続ける自動人形のアームが不意に止まる
PRINTFORMW すっかり打ちへしがれた%QST_LEADER_NAME()%が恐る恐る見上げると、可動部から煙を上げて自動人形が停止していた
PRINTFORMW どうやら老朽化した部品が自身の激しい稼働に耐えられなかったらしい
PRINTL
PRINTFORML なんとか拘束を解き、%QST_LEADER_NAME()%は痛む身体を持ち上げる
PRINTFORMW 手ひどく消耗してしまったが命はまだある……このまま探索を続けるなら一度身体を癒す必要がありそうだ……
CALL QST_CHANGE_ITEM("淫堕点,10,")

;-------------------------------------------------
;EV105
;モブエンカウント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_105
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_105

CALL HO_GET_ROOM_NAME(HO_MASTER現在地)
LOCALS = %RESULTS%
SELECTCASE RAND:3
    CASE 0
        CALL PRINT_NOTICE(@"敵襲だ！",,1)
        PRINTFORMW %QST_LEADER_NAME()%は武器を握りしめ、敵に対峙した
    CASE 1
        CALL PRINT_NOTICE(@"%LOCALS%では魔物が待ち構えていた！",,1)
        PRINTFORMW 突破するには眼前の驚異を打ち払わねばならない
    CASEELSE
        CALL PRINT_NOTICE(@"%QST_LEADER_NAME()%が%LOCALS%に足を踏み入れるなり魔物が襲いかかってきた！")
ENDSELECT

CALL QST_CEV_COMBAT("墳墓")
IF !RESULT
    CALL QST_QUEST_11_DEFEAT_EVENT()
ENDIF
CALL QST_DEACTIVATE_THIS_EVENT()

;-------------------------------------------------
;EV100 施錠された宝箱
;EV101 宝箱
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_100
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME
#DIM  CHEST_TYPE
EVENT_SLOT_NAME '= @"%S_R("淫らな意匠の小箱/施錠された宝箱/裸像の抱える宝飾箱")%"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_100
#DIMS DYNAMIC _ITEM_LIST
#DIM  DYNAMIC CHEST_TYPE

CALL QST_CEV_CREATE_RANDOM_CHEST(85, 10, 5)
SELECTCASE RESULT
    CASE 0
        SELECTCASE RAND:4
            CASE 0
                _ITEM_LIST '= @"MONEY,2000,"
            CASE 1
                _ITEM_LIST '= @"{ITEM_ポーション},1,"
            CASE 2
                _ITEM_LIST '= @"{ITEM_鋼インゴット},2,"
            CASE 3
                _ITEM_LIST '= @"{ITEM_魔獣の肉片},3,"
        ENDSELECT
        CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST, 100, 100, "LEWD_LV2")
    CASE 1
        _ITEM_LIST '= @"{ITEM_銀インゴット},2,"
        CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST, 100, 130, "LEWD_LV2")
    CASE 2
        _ITEM_LIST '= @"{ITEM_金インゴット},1,"
        CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST, 100, 130, "LEWD_LV2")
ENDSELECT

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
@QST_QUEST_11_DEFEAT_EVENT_2
#DIM CHARA_ID
#DIM 口上カラー_カール
#DIM 口上カラー_カーラ

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
IF CHARA_ID == -1
    CALL QST_QUEST11_CREATE_UNIQUE_MOB1
    CHARA_ID = RESULT
ENDIF
口上カラー_カール = CFLAG:CHARA_ID:口上カラーコード
CFLAG:CHARA_ID:面識 = 1

CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
IF CHARA_ID == -1
    CALL QST_QUEST11_CREATE_UNIQUE_MOB2
    CHARA_ID = RESULT
ENDIF
口上カラー_カーラ = CFLAG:CHARA_ID:口上カラーコード
CFLAG:CHARA_ID:面識 = 1

PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
PRINTFORMW 気を失ってしまった%QST_LEADER_NAME()%が意識を取り戻すと、二人の少年、少女が顔を覗き込んでるのを認めた
PRINTFORMW 一見は人間の子どものようだが、背中から生えた翼は彼らが人ならざる者であることを物語っている
PRINTL
CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あちゃー、今回もやりすぎちゃったね。カール」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うーん……今の人間は昔より随分脆いなぁ……。もう少し難易度を下げた方がいいのかなぁ……」@/C@", 1)
PRINTW
PRINTL
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「おっと、ごめんごめん、目が覚めたみたいだね。ニンゲン」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「詳しいことは省くけど、僕らは淫魔、この迷宮の主だ」@/C@", 1)
WAIT
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「君は無様にも迷宮の途中で落伍した。本来ならここで死んでも文句は言えない」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「でもね、慈悲深ーい私たちは、アナタにチャンスをあげようと思うの！ちょっとばかりお礼はいただくけどね！」@/C@", 1)
PRINTL
PRINTFORMW %QST_LEADER_NAME()%は一体何なにをさせるつもりだと淫魔に問う
PRINTL
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「どうしたの？ただでさえ鈍い頭を打っていよいよ駄目になったの？淫魔の要求なんて一つしかないだろうに」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「これから、あなたのこと散々犯して！辱めて！玩具にするの！たったそれだけ！」@/C@", 1)
PRINTW
PRINTL
PRINTFORMW %QST_LEADER_NAME()%は傲慢で一方的な二人の言に言葉を絞り出して抗議する
PRINTL
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「あー、ごめん、なにか言った？豚の言葉ってどうにも聞き取りにくくてね。まぁ多分死なないから、ちょっと我慢してよ」@/C@", 1)
PRINTW
CALL QST_QUEST_11_DEFEAT_EVENT_3

@QST_QUEST_11_DEFEAT_EVENT_3
#DIM CHARA_ID

FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 20

;主導除外フラグが立ったキャラクターは慰安調教で常にターゲットに回る。
;拘束中など自主的に行動させたくないシチュエーションでこのフラグを立てる。
CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):凌辱被害者フラグ = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):調教参加フラグ = 1
NEXT

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
IF CHARA_ID != -1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:調教参加フラグ  = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
ELSE
    CALL QST_QUEST11_CREATE_UNIQUE_MOB1
    CFLAG:RESULT:調教参加フラグ  = 1
    CFLAG:RESULT:強制友好化 = 1
    CFLAG:RESULT:慰安参加者 = 1
    CFLAG:RESULT:慰安参加者非対象化 = 1
ENDIF
CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
IF CHARA_ID != -1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:調教参加フラグ  = 1
    CFLAG:CHARA_ID:強制友好化 = 1
ELSE
    CALL QST_QUEST11_CREATE_UNIQUE_MOB2
    CFLAG:RESULT:調教参加フラグ  = 1
    CFLAG:RESULT:強制友好化 = 1
    CFLAG:RESULT:慰安参加者 = 1
    CFLAG:RESULT:慰安参加者非対象化 = 1
ENDIF
CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(21)

PRINTL
CALL PRINT_TITLE("双子淫魔の搾精凌辱", "エロイベント")
WAIT
RETURN

;-------------------------------------------------
;EV15 カール&カーラ -調教後
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_21
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_11_EVENT_21
#DIM CHARA_ID
#DIM 口上カラー_カール
#DIM 口上カラー_カーラ

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
口上カラー_カール = CFLAG:CHARA_ID:口上カラーコード
CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
口上カラー_カーラ = CFLAG:CHARA_ID:口上カラーコード

PRINTFORMW 二人の淫魔のあまりにも嗜虐的な責めに%QST_LEADER_NAME()%はもはや微かなうめき声をあげることしか出来ないでいる……
PRINTL
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ちょっと……だめだよ、カーラ。それ以上やったら。それ、多分もう死んじゃうから」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「危ない危ない、食べ尽くしちゃうところだったよ……！」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「今日のところはこれくらいにしておいてあげるね。次はもう少し分別を持って遊びに来てね！」@/C@", 1)
PRINTW
CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「……でないと、次は本当に喰い殺すよ？」@/C@", 1)
WAIT
CALL QST_QUEST_11_DEFEAT_EVENT

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
@QST_QUEST_11_DEFEAT_EVENT
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL

PRINTFORMW あれからどうやって%QST_LEADER_NAME()%が最寄りの街まで帰還したのか、記憶が酷く混濁しており定かではない
PRINTFORMW 気がつけば宿のベッドに身を横たえており、強烈な痛みと重苦しい倦怠感に苛まれ、数日は動くことも叶わなかった
PRINTFORML
PRINTFORMW ……なぜあの迷宮を探索した冒険者たちが多くを語ろうとしないのか、身を持って理解することができた
PRINTL
PRINTFORMW ……もし次があるならば、あのインキュバスの言の通り、より一層の注意を払うべきだろう……
PRINTL

CALL QST_CHANGE_ITEM("淫堕点,10,")
CALL PRINT_NOTICE("治療費や宿泊費として10,000ディナリを失った……")
MONEY -= 10000
TFLAG:クエスト_完了フラグ = 1
QST_COOLTIME:QST_有効クエストID += 4
CALL TRAIN_EXIT()

;-------------------------------------------------
;慰安モブの定義
;-------------------------------------------------
@QST_QUEST11_CREATE_MOB(SEX, MOB_NAME)
#DIM SEX
#DIMS MOB_NAME
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)
FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
CALLNAME:対象 = %名前%
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
TALENT:対象:Ｃ鈍感 = 1
ABL:対象:戦闘 =  RAND(35, 50)
ABL:対象:性知識 = 3
ABL:対象:奉仕 = 1
ABL:対象:性技 = RAND(0, 4)
ABL:対象:欲望 = RAND(3, 7)
ABL:対象:性交 = RAND(3, 5)
ABL:対象:射精 = 0
ABL:対象:サド = RAND(0, 1)
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = RAND(0, 300)
CFLAG:対象:調教参加フラグ = 1
CFLAG:対象:慰安モブ = 1

LOCALS = %S_R("male01/shota01/shota02/shota03")%
CSTR:対象:顔グラファイル名 '= @"%LOCALS%.webp"
CSTR:対象:キャラ顔グラ '= @"%LOCALS%.webp"
CFLAG:対象:ポートレイト画像タイプ = 2

RETURN 対象

;-------------------------------------------------
;クエスト用ユニークキャラクタ
;-------------------------------------------------
@QST_QUEST11_CREATE_UNIQUE_MOB1()
#DIM SEX = 0
#DIMS MOB_NAME = "カール"
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)

FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
;ユニークキャラとして一度だけ生成するのでCSTR:99を登録
CSTR:対象:99 '= "インキュバス・カール"
CALLNAME:対象 = %名前%
BASE:対象:体力 = 7500
BASE:対象:気力 = 5500
MAXBASE:対象:体力 = 7500
MAXBASE:対象:気力 = 5500
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
TALENT:対象:絶倫 = 1
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
TALENT:対象:成長型 = 成長型_戦士
ABL:対象:戦闘 =  120
ABL:対象:魔術 =  150
ABL:対象:性知識 = 5
ABL:対象:奉仕 = 4
ABL:対象:性技 = 6
ABL:対象:欲望 = 10
ABL:対象:性交 = 15
ABL:対象:射精 = 15
ABL:対象:サド = 10
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = 500
CFLAG:対象:調教参加フラグ = 1

LOCALS '= "incubus01"
CSTR:対象:顔グラファイル名 '= @"%LOCALS%.webp"
CSTR:対象:キャラ顔グラ '= @"%LOCALS%.webp"
CFLAG:対象:ポートレイト画像タイプ = 2
CFLAG:対象:汎用口上パターン = 2000
CFLAG:対象:口上カラーコード = 0xff6666
CFLAG:対象:所属 = 所属_ゲストNPC

RETURN 対象

@QST_QUEST11_CREATE_UNIQUE_MOB2()
#DIM SEX = 1
#DIMS MOB_NAME = "カーラ"
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)

FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
;ユニークキャラとして一度だけ生成するのでCSTR:99を登録
CSTR:対象:99 '= "サキュバス・カーラ"
CALLNAME:対象 = %名前%
BASE:対象:体力 = 7500
BASE:対象:気力 = 5500
MAXBASE:対象:体力 = 7500
MAXBASE:対象:気力 = 5500
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
TALENT:対象:絶倫 = 1
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
TALENT:対象:成長型 = 成長型_詩人
ABL:対象:戦闘 =  120
ABL:対象:魔術 =  150
ABL:対象:性知識 = 5
ABL:対象:奉仕 = 4
ABL:対象:性技 = 6
ABL:対象:欲望 = 10
ABL:対象:性交 = 15
ABL:対象:射精 = 15
ABL:対象:サド = 10
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = 500
CFLAG:対象:調教参加フラグ = 1

LOCALS '= "succubus01"
CSTR:対象:顔グラファイル名 '= @"%LOCALS%.webp"
CSTR:対象:キャラ顔グラ '= @"%LOCALS%.webp"
CFLAG:対象:ポートレイト画像タイプ = 2
CFLAG:対象:汎用口上パターン = 2001
CFLAG:対象:口上カラーコード = 0x40C0FF
CFLAG:対象:所属 = 所属_ゲストNPC

RETURN 対象


code:
;-------------------------------------------------
;EV9 インキュバスの扉
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_20
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_SLOT_NAME = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_20
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID
#DIM 口上カラー_カール
#DIM 口上カラー_カーラ
#DIM 評定額

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
IF CHARA_ID == -1
    CALL QST_QUEST11_CREATE_UNIQUE_MOB1
    CHARA_ID = RESULT
ENDIF
口上カラー_カール = CFLAG:CHARA_ID:口上カラーコード
CFLAG:CHARA_ID:面識 = 1

CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
IF CHARA_ID == -1
    CALL QST_QUEST11_CREATE_UNIQUE_MOB2
    CHARA_ID = RESULT
ENDIF
口上カラー_カーラ = CFLAG:CHARA_ID:口上カラーコード
CFLAG:CHARA_ID:面識 = 1

;初回
IF !QST_SV_QUEST_FLAG:QST_有効クエストID
    PRINTFORML 長い回廊を抜け、%QST_LEADER_NAME()%は突き当りの扉に立った。恐らくここが最後の部屋であろう
    PRINTFORMW 意を決して扉を開けるとまず目の前に飛び込んできたのは乱交と痴態に耽る無数の肉のうねり……を模した像の数々だ
    PRINTL
    PRINTFORMW 交わり、絡み合う肢体の彫像の山の上には少年と少女が腰掛けている……
    PRINTFORMW 一見はあたかも双子のようだ。ただし少年の方は赤褐色の瞳、少女の方は冷たい青い瞳をしている
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「おっ、ここまで来られた人間は本当に久しぶりだよ。ようこそ淫魔の間へ」@/C@", 1)
    PRINTL
    PRINTFORMW 少年が気さくに語りかけてくる
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あはは、アナタが頑張ってるところここからじっくり見させてもらってたよ！」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「必死になってるニンゲンって……やっぱり見てて本当に面白いね！」」@/C@", 1)
    PRINTW
    PRINTFORML 少女の小馬鹿にしたような態度に%QST_LEADER_NAME()%は苛立ちを覚える
    PRINTFORMW 見た目こそ人間の子どもだが、背中から生えた翼や容姿から、二人は恐らく魔族、淫魔の類だろう……
    PRINTL
    PRINTFORMW 同時に……酷く恐ろしい悪寒が背筋を伝うのを%QST_LEADER_NAME()%は感じた
    PRINTL
    PRINTFORMW 直感は、この二人と剣を交えたところで絶対に勝利し得ないことを告げている……
    PRINTL
    CALL PRINT_TITLE("双子の淫魔", "イベント")
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ふ～ん。結構勘はいいんだね。僕はカール。インキュバスのカールだ」@/C@", 1)
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「こっちはサキュバスのカーラ。僕たちはこの迷宮の主でね、二人とも昔は結構崇められたり、奉られたりしたんだよ」@/C@", 1)
    PRINTW
    PRINTFORML インキュバス、サキュバス二人の淫魔は軽やかな足取りで間合いを詰めてくる
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「昔はニンゲンが色々と捧げてくれたんだよね～。可愛い娘とか男の子とか」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「その代わりに私たちはちょっと魔力を分けてあげたり、ニンゲンがありがたがっている石ころをあげたりしたんだ」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「その辺に転がってるのはみ～んなその名残。絞り尽くしちゃって、イケニエちゃんはみんな空っぽの石ころになっちゃった」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「最後にイケニエちゃん貰ってからもう何百年になるのかな～、最近また少し賑やかになってきたけど……長い間すっごく暇だったんだよ！」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「さて、そんなわけで人間！取引といこう」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「僕たちは別に今君を取って食おうとは思っていない」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ただ君がここに来るまでに味わい、散々溜め込んできた\"欲\"をちょっと分けてくれさえすればいい」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「代わりに僕らは君の役に立つものを提供しよう。どうだい？悪い取引ではないだろう？」@/C@", 1)
    WAIT
    PRINTL
    PRINTFORML 淫魔たちの提案は、実質的に選択肢が無いように思われた
    PRINTFORMW 断れば……恐らくただでは済まないだろう。さて、どうしたものか
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1
ELSE
    PRINTFORML 長い回廊を抜け、%QST_LEADER_NAME()%は突き当りの扉に立った。扉の向こうからは並々ならぬ気配がする
    PRINTFORMW 意を決して再び扉を開けると二人の淫魔が挑発的な笑みを浮かべてこちらを眺めていた……
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あは！また遊びに来てくれたんだね！ぐちゃぐちゃにされるのが癖になっちゃったのかな？可愛いね！」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「まぁ楽しんでくれているなら僕たちも苦労した甲斐があったというものだよ」@/C@", 1)
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「さて、それじゃあ取引と行こうか。もちろん君の色欲を分けて……くれるんだよね？」@/C@", 1)
ENDIF

PRINTL
CALL ASK_YN("取引を受ける", "抵抗する")
PRINTL

IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%が渋々頷くとカールと名乗る淫魔はにこりと微笑んだ
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うんうん、話が早くて助かるよ。君は中々活きがいいからね。僕たちとしても大事にしたいんだ」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「釣りは釣果よりも、釣る行為自体が面白いだろう？だから、長く楽しめるようにしないとね」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「キャッチアンドリリースってやつだね！」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「それじゃあ早速欲情を分けてもらおうか……、どれどれ……」@/C@", 1)
    PRINTL
    CALL PRINT_TITLE("淫魔の品定め", "エロイベント")
    PRINTW
    CALL QST_QUEST_11_EVENT_20_淫魔の評定
    評定額 = RESULT
    PRINTL
    PRINTFORMW …………
    SELECTCASE 評定額
        CASE IS >= 100
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うんうん！イイね！すごくいい、本当に最上級のエロ雌豚だ！」@/C@", 1)
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「君のみっともなくてどうしようもない痴態を眺めてるだけでも、僕たちはもうしばらくやっていけそうだよ！」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「エロ雌豚ちゃんはもう頭の中真っピンクでエッチなことしか考えられないって感じだね！可愛いね！」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「豚ちゃんはきっと普通にエッチするだけじゃ満足できない子だろうから、またいつでも遊びに来てね！」@/C@", 1)
            WAIT
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「その時はもしかしたら～、今度は私達が直接遊んであげるかも！」@/C@", 1)
            PRINTW
        CASE IS >= 30
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うん。まぁ、まぁまぁかな。中の上くらいの豚」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「お姉さんはぁ……ちょ～っと残念かな。もっとアナタがヒィヒィ言いながらエロ汁垂れ流して走り回ってるところ見たかったんだけどなぁ～」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「でもちょっとは面白かったから、これ、あげるね！次はもうちょっとがんばってね！」@/C@", 1)
            PRINTW
        CASEELSE
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「はぁ……、あのさ……全然ダメじゃん。やる気あるの？なんていうかさ、不真面目すぎるんだよね……」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「そうだねぇ……カール。これは期待外れ……。ちょっと間違えて殺しちゃいそう」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「……一応は来てくれたお客さんだからこれあげるけどさ、次はもうちょっと、真面目にやってね」@/C@", 1)
            PRINTW
    ENDSELECT
    PRINTL
    PRINTFORMW 淫魔たちは%QST_LEADER_NAME()%を散々好き放題に罵倒した上で、約束通り報酬を手渡してきた……
    PRINTFORMW %QST_LEADER_NAME()%は羞恥に顔を赤らめ、歯ぎしりつつも、剣を交えずとも実感できる圧倒的な力量差を前に今はただ去ることしかできなかった……
    PRINTL
    PRINTFORMW 一先ず調査は完了した……。帰還するとしよう
    PRINTL
    CALL QST_CHANGE_ITEM(@"淫堕点,{MIN(評定額 / 10 * 10, 200)},")
    SELECTCASE 評定額
        CASE IS >= 100
            CALL QST_CHANGE_ITEM(@"{ITEM_金インゴット},1,")
            CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},3,")
        CASE IS >= 30
            CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},3,")
        CASEELSE
            CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},1,")
    ENDSELECT
    TFLAG:クエスト_完了フラグ = 1
    QST_COOLTIME:QST_有効クエストID += 4
    QST_SV_QUEST_FLAG:QST_有効クエストID = 2
    CALL TRAIN_EXIT()

ELSE
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「……やれやれ、昔からまぁ馬鹿な連中だとは思ってたけど、何百年たってもちっとも変わらないね……」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「ねぇねぇカール、身の程知らずの家畜は～えっと、今風に言うなら……『わからせ』が必要だよね！」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「そうだね、カーラ。少し痛めつけて、己の立場をきちんとわきまえさせないといけないね」@/C@", 1)
    PRINTW
    PRINTFORMW 二体の淫魔がその身に秘めていた膨大な魔力を解き放つ……
    PRINTFORMW 空間がビリビリと震え、周囲の彫像が砕けて弾け飛ぶ。発狂しかねないほどの圧力と恐怖が%QST_LEADER_NAME()%を苛む
    PRINTL
    PRINTFORMW 解き放たれた膨大な魔力は、禍々しい黒馬の姿をとって顕現し、%QST_LEADER_NAME()%を包囲する
    PRINTL
    PRINTFORMW 状況は絶望的だ……。しかし後悔したところで最早手遅れだろう。圧力に負けて命までは手放さぬよう、死力を尽くして抗わねば……
    CALL CC_INIT()
    CALL CC_ADDCHARA("DEFENDER", NAME_TO_CHARA("インキュバス・カール"))
    CALL CC_ADDCHARA("DEFENDER", NAME_TO_CHARA("サキュバス・カーラ"))
    CALL CC_ADDMOB("DEFENDER", "ナイトメア１", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)
    CALL CC_ADDMOB("DEFENDER", "ナイトメア２", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)
    CALL CC_ADDMOB("DEFENDER", "ナイトメア３", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)
    CALL CC_ADDMOB("DEFENDER", "ナイトメア４", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)

    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
    NEXT
    CALL CLOSE_COMBAT()
    ;今のところ負けイベ
    IF RESULT == 2
        PRINTFORMW 激烈な魔力の奔流が炸裂し、%QST_LEADER_NAME()%は抗う余地もなく地面に叩きつけられた！
        PRINTL
        PRINTFORMW 朦朧とする意識の中、二人の淫魔が嘲笑の笑みを浮かべて自身を睥睨している
        PRINTL
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「ぷっ……アハハ！ダメじゃん！全然ダメじゃん！準備運動にもならないよ！」@/C@", 1)
        PRINTW
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「まぁ威勢の良さに免じて命は取らないでおいてあげるよ。でもね、聞き分けのない家畜は……ちゃんと躾けないとね」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「大丈夫？意識はある？これから好き勝手に虐めて絞らせてもらうけど、死なないように頑張ってね！」@/C@", 1)
        PRINTW
        CALL QST_QUEST_11_DEFEAT_EVENT_3()
    ELSE ;勝てないはず
        PRINTL
        PRINTFORMW %QST_LEADER_NAME()%は圧倒的な力で黒馬を屠り、淫魔を真正面からねじ伏せた……
        PRINTL
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「えっ、ちょっとまって？なんで？なんでこの状況で勝てるの？」@/C@", 1)
        PRINTW
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「困ったなぁ……君みたいな規格外なのが来ることは考えてなかったよ……」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「あーもう、参ったなぁ……こうなったときの段取り、なにも用意してないんだよね……」@/C@", 1)
        PRINTW
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「とりあえずさ、これあげるからさ……。今日は帰ってくれない？ごめんね！」@/C@", 1)
        PRINTW
        CALL QST_CHANGE_ITEM("名誉点,150,")
        PRINTFORMW 当初の目算が外れたのか、淫魔達はバツが悪そうに顔を見合わせると、瞬く間に姿を消してしまった
        PRINTL
        PRINTFORMW 一先ず調査は完了した……。帰還するとしよう
        PRINTL
        TFLAG:クエスト_完了フラグ = 1
        QST_COOLTIME:QST_有効クエストID += 4
        QST_SV_QUEST_FLAG:QST_有効クエストID = 2
        CALL TRAIN_EXIT()

    ENDIF
ENDIF

;-------------------------------------------------
; 探索の質を評価。ついでに罵倒される
;-------------------------------------------------
@QST_QUEST_11_EVENT_20_淫魔の評定
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID
#DIM 口上カラー_カール
#DIM 口上カラー_カーラ
#DIM DYNAMIC ぶっかけ度
#DIM DYNAMIC 評定額

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
口上カラー_カール = CFLAG:CHARA_ID:口上カラーコード

CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
口上カラー_カーラ = CFLAG:CHARA_ID:口上カラーコード

CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, 0))
;ぶっかけ具合
FOR LOCAL, 0, CUM_SLOT_NUM
    ぶっかけ度 += CUM_GET_AMOUNT:CHARA_ID:LOCAL
NEXT
ぶっかけ度 = ぶっかけ度 / 3
IF ぶっかけ度 > 35
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うわぁ……身体中どろどろのザーメンまみれじゃん……。人間の雌豚の盛り方って昔から品がないないなぁ……」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「まぁ雌豚は雌豚らしく穴からザーメン垂れ流しながらアクメ顔晒してるくらいの方がそれらしくていいと思うよ」@/C@", 1)
    評定額 += 25
    PRINTW
ELSEIF ぶっかけ度 > 15
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「いい感じに身体から青臭い精臭撒き散らしてるね♪」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「香水なんかよりずっと魅力的だと思うよ！定期的にぶっかけてもらったらいいんじゃないかな！」@/C@", 1)
    評定額 += 10
    PRINTW
ENDIF
DEBUGPRINTFORML ぶっかけ度:{ぶっかけ度}
;露出_胸
IF C_COND("露出_胸:露出",, CHARA_ID)
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あはっ！お洋服、破けちゃったんだ。可愛いおっぱい丸出しにしてここまで歩いてきたの？恥ずかしくないの？」@/C@", 1)
    評定額 += 25
    PRINTW
ENDIF
;露出_秘部
IF C_COND("露出_秘部:露出",, CHARA_ID)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「下着は……はいてないの？それともグチョグチョにしちゃったから脱いできたの？」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「恥ずかしいところ丸見えじゃん。それってどうなのかなぁ……？」@/C@", 1)
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ニンゲンって、一応は社会性とかあるんだよね？」@/C@", 1)
    評定額 += 25
    PRINTW
ENDIF
;欲情LV
LOCAL = GET_PALAMLV(PALAM:(CHARA_ID):欲情)
SELECTCASE LOCAL
    CASE IS >= 12
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「大丈夫？息荒いよ？こんなに顔真っ赤にして、息も荒らげて……ただでさえミニマムな頭のなかを……スケベなことでいっぱいにしてるの？」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「うんうん、いいよ！すごくいいと思う！余計なことなんて考えないで、どうやって気持ちよくなるかだけ考えてる方が絶対楽しいよ！」@/C@", 1)
        PRINTW
    CASE IS >= 8
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「ふーん……。まだちょっと物足りないって顔してるね。まぁ余計なこと考えないで、もっと頭真っピンクにした方が楽しいと思うよ！」@/C@", 1)
        PRINTW
    CASEELSE
ENDSELECT
評定額 += LOCAL * 3
;潤滑LV
LOCAL:0 = GET_PALAMLV(PALAM:(CHARA_ID):潤滑)
SELECTCASE LOCAL
    CASE IS >= 9
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「アハハ！お股もヌルヌルでグショグショじゃん！きっといっぱいイッちゃったんだろうねぇ……」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「……どう、気持ちよかった？ねぇ、人様のおうちでメス汁ぽたぽた垂れ流して、汚しながら無様に喘いで、のたうち回って……、気持ちよかったんだよね？」@/C@", 1)
        PRINTW
    CASE IS >= 6
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「わぁ、お股びちょびちょだねぇ……。でもまだちょっと濡れ足りないのかな？」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「雌豚は潮吹いて浅ましくイキ散らかしてるくらいの方が、可愛げがあっていいと思うんだ！」@/C@", 1)
        PRINTW
    CASEELSE
ENDSELECT
評定額 += LOCAL * 3
RETURN 評定額

repo_name: pervysquirrelz/erafl

path: ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮/QUEST_11_淫魔の小迷宮_淫魔の間.ERB

license: unknown


code:
MAP_NAME = "淫魔の小迷宮"
MAP_IMAGE = "淫魔の小迷宮.webp"
PATH_COLOR = "0xff00"
PATH_ALPHA = "0x0"
NODE_BASE_COLOR = "0xEE1190"

path: ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮/淫魔の小迷宮.txt


code:
<DocumentElement>
  <GMAPDATA>
    <id>638145913543188710</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>入口</NODE_NAME>
    <POS_X>100</POS_X>
    <POS_Y>360</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913590166636</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>彫像の間</NODE_NAME>
    <POS_X>140</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>0,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913635016064</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>1,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913734256553</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>大部屋</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>2,4,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913755294270</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>60</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913907045903</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>3,6,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913924505656</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>書斎</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>60</POS_Y>
    <PATH_LIST>5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913945666133</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>140</POS_Y>
    <PATH_LIST>5,8,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145914087637586</id>
    <NODE_ID>8</NODE_ID>
    <NODE_NAME>通路</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>7,9,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145914111958370</id>
    <NODE_ID>9</NODE_ID>
    <NODE_NAME>淫魔の間</NODE_NAME>
    <POS_X>500</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>8,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮/淫魔の小迷宮.xml


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_12:野盗討伐
;-------------------------------------------------
@QST_TITLE_QUEST_12
RESULTS '= "野盗討伐"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_12
RESULTS '= "周期,探索,"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_12
RESULTS:0 '= "野盗の討伐はギルドに寄せられるもっとも一般的な依頼のひとつだ"
RESULTS:1 '= "盗賊の一団は古い山腹の遺跡を根城にしている。手配されている賊の頭を討ち取れば依頼達成となる"
RESULTS:3 '= "帝国が霧散し、大陸横断的な国家権力が消滅して以来、辺境地域は無法地帯と化している"
RESULTS:4 '= "自前の戦力を仕立てる余力のない者は冒険者や傭兵を頼ったが、"
RESULTS:5 '= "一方で食いっぱぐれた冒険者や傭兵が野盗に転向するケースも少なくはない"
RESULTS:6 '= "各地で諸勢力が勃興した今日においても、混沌とした辺境の情勢は昔とさほど変わらないようだ"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_12
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★★@/C@"
RESULTS:1 '= "報酬：6,000ディナリ～,求人書類(初回のみ)"
RESULTS:2 '= "推奨パーティ人数:3人以上　　　推奨平均戦闘値：40以上"
RESULTS:3 '= "備考:定期性クエスト(4期毎)"
RESULTS:4 '= @"@C:{カラー_選択不可}@「冒険者も傭兵も盗賊もその日暮らしの流れ者である点には変わらん」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「コインの裏表みてぇなもんだし、コロコロ向きを変える奴もごまんといる」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「おまえさんはどうなんだろうな。まぁ俺の知ったことではないが」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_12(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_0_ゴブリンの洞窟の攻略が必要
SIF !QST_SV_QUEST_FLAG:0
    RETURN 0
SIF QST_COOLTIME:QUEST_ID
    RETURN 0
RETURN 1

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_12(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_12_野盗討伐"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "mapdata"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,1,]"
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,2,]"
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,4,100,101,102,]"
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,6,7,]"
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,5,]"
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,8,]"

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_12
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_12

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_12_ABORT
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_12
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL

PRINTFORML %QST_LEADER_NAME()%は手配書の示す盗賊団の拠点に足を向けた
PRINTFORMW どうやら連中は古い時代の遺跡をねぐらにしているようだ
PRINTL
PRINTFORML 敵の戦力はさほど大きくはないようだが油断は禁物だ
PRINTFORMW 思わぬ罠が待ち受けている可能性もある
PRINTL
PRINTFORMW 常に警戒を怠るべきではないだろう……

FLAG:クエスト_中断を許可 = 0
;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 1
        CALL HO_PRINT_LOG("上古の遺跡のようだ。現代とも古帝国とも異なる、異質な様式の空間が広がっている", 1)
    CASE 2
        CALL HO_PRINT_LOG("部屋の中は青臭い、むせ返るような精臭と汗の臭いで満たされている", 1)
    CASE 3
        CALL HO_PRINT_LOG("盗賊が近隣から略奪した品々が部屋の隅に乱雑に積み上げられている", 1)
    CASE 5
        CALL HO_PRINT_LOG("天井には無数の配管が渡されており、壁には奇怪な装置の数々が並んでいる", 1)
    CASE 6
        CALL HO_PRINT_LOG("武器庫のようだ。粗野な作りの武器が乱雑に立て掛けられている", 1)
ENDSELECT

;-------------------------------------------------
;EV1
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_1
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_1
#DIM EVENT_ID = 1
PRINTFORMW 日が暮れるを待ち%QST_LEADER_NAME()%は野盗が拠点にしている山腹の遺跡に忍び込んだ
PRINTFORMW 盗賊団はさほど規模も大きくないのか警戒は手薄なようで、内部への侵入は難なく成功した
PRINT
PRINTFORML エントランスからは南北に二本の通路が伸びている
PRINTFORMW さて、どちらへ進もうか

;-------------------------------------------------
;EV2
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_2
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_2
#DIM CONST EVENT_ID = 1
#DIM CONST 派生イベントID = 3
#DIM EVENT_FLAG
#DIM CHARA_ID
CALL QST_GET_EVENT_FLAG(派生イベントID)
EVENT_FLAG = RESULT
;このイベントは後続の派生イベント(EV2)の選択結果で2回目以降の処理が変わるのでセットされたフラグを見て内容を分岐
SELECTCASE EVENT_FLAG
        ;初回
    CASE 0
        PRINTFORMW 小部屋の奥からくぐもった嬌声と水音が聞こえる……
        PRINTFORMW 陰から様子を伺ってみると、どうやら野盗どもが"戦利品"を弄んでいるようだ
        ;調教モードをクエスト -> 慰安に切替
        FLAG:ウフフフラグ = 1
        FLAG:調教モード = 調教_慰安
        TFLAG:調教ターン数 = 8
        FLAG:調教スキップ可 = 1

        CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
        ;竿役生成
        FOR LOCAL, 0, 3
            CALL QST_QUEST_12_CREATE_RAPER(0)
            CHARA_ID = RESULT
            CFLAG:CHARA_ID:調教参加フラグ = 1
            CFLAG:CHARA_ID:慰安参加者 = 1
            CFLAG:CHARA_ID:強制友好化 = 1
            CFLAG:CHARA_ID:慰安参加者非対象化 = 1
        NEXT
        ;適当に被害者を追加
        FOR LOCAL, 0, 4
            CALL QST_QUEST_12_CREATE_TRAIN_VICTIMS
        NEXT
        ;調教後に呼び出される後続イベント
        CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
    CASE 1
        PRINTFORMW 小部屋には狼藉者どもの死体が散乱している
        ;被害者を見捨てる
    CASE 2
        PRINTFORMW 野盗どもは欲望の赴くままに陵辱を続けている……
        PRINTFORMW 女の肢体を弄ぶのに夢中なのか、潜んでいる%QST_LEADER_NAME()%に気付く様子はない
ENDSELECT

;-------------------------------------------------
;EV3
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_3
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_3
#DIM CONST EVENT_ID = 3
#DIM EVENT_FLAG
PRINTFORMW 野盗どもは女を凌辱するのに夢中になっており、%QST_LEADER_NAME()%に気付く様子はない……
PRINTFORMW 彼女らを救出すべきだろうか……？
CALL COLOR_PRINTW("ここで騒ぎを起こせば敵の注意を引くかもしれない……", カラー_注意)
CALL ASK_YN("救出を試みる", "見捨てる")
IF RESULT == 0
    PRINTFORMW %QST_LEADER_NAME()%は隙だらけの狼藉者どもを強襲した！
    ;戦闘イベント
    CALL CC_INIT()
    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
    NEXT
    CALL CC_ADDMOB("DEFENDER", "ならず者１", 2000, 2000, 30)
    CALL CC_ADDMOB("DEFENDER", "ならず者２", 2000, 2000, 30)
    CALL CC_ADDMOB("DEFENDER", "ならず者３", 2000, 2000, 30)
    CALL CLOSE_COMBAT()
    ;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
    IF RESULT == 1
        PRINTFORML 戦闘に勝利した%QST_LEADER_NAME()%は女性たちの解放に成功した
        PRINTFORMW 彼女らは涙ながらに礼を告げると、覚束ない足取りで出口へと向かった
        PRINTFORMW 謝礼として幾ばくかの金品を得た
        CALL COLOR_PRINTW("3,500ディナリを得た", カラー_注意)
        MONEY += 3500
        CALL COLOR_PRINTW(@"――　名誉点を15点得た！", カラー_注意)
        QST_PRESTIGE_POINT += 15
        CALL CC_ADD_COMBAT_EXP(4)
        CALL CC_APPLY_CHARA_HP()
        WAIT
        CALL QST_SET_EVENT_FLAG(EVENT_ID, 1)
    ELSE
        CALL CC_APPLY_CHARA_HP()
        CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%はならず者に返り討ちにされてしまった……", カラー_注意)
        PRINTFORMW 「ははっ、こいつぁは傑作だ。俺たちに勝てるとでも思ったのか？この間抜けめ」
        PRINTFORMW 「いまここでぶっ殺しちまってもいいが……まぁせっかくの新しい玩具だ、精々楽しませてもらうとしよう」
        PRINTFORMW 賊は敗北した%QST_LEADER_NAME()%を拘束すると、下卑た笑みを浮かべて弄びはじめた……
        CALL QST_QUEST_12_DEFEAT_EVENT
    ENDIF
ELSE
    PRINTFORMW 無用の危険を冒すべきではないと判断した%QST_LEADER_NAME()%はそっとその場を後にした……
    CALL QST_SET_EVENT_FLAG(EVENT_ID, 2)
ENDIF

;-------------------------------------------------
;EV4　倉庫
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_4
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_12_EVENT_4

PRINTFORML ここはどうやら山賊たちがかき集めた略奪品を収蔵している倉庫のようだ
PRINTFORMW なにか使えるものはないか物色してもよいし、そのまま奥へと進んでもよい
PRINTL
PRINTFORMW 奥へ進むなら、左か右か、どちらか選ぶ必要がある
PRINTL
PRINTFORMW 東側の小部屋は武器庫のようだ。中からは微かに人の気配がする……
PRINTL
PRINTFORML 東側の敵は強固に武装している可能性が高い
PRINTFORMW 先へ進むならまずは西側の部屋に忍び込み様子を見るのが賢明だと%ANAME(QST_RANDCHARA())%は思案した
PRINTL
PRINTFORMW ……さて、どちらに歩みを進めようか

CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV5 東の小部屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_5
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_5
#DIM EVENT_FLAG

CALL QST_GET_EVENT_FLAG(5)
IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%が部屋に入るや、中にいた賊は即座に武器を手に取り応戦の構えを取った
    PRINTFORML 元は傭兵か冒険者だったのだろう、何名かは明らかに場馴れしている様子だ
    PRINTFORMW 厄介な戦いになるかもしれない……
    ;戦闘イベント
    CALL CC_INIT()
    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
    NEXT
    CALL CC_ADDMOB("DEFENDER", "傭兵くずれ１", 2500, 2500, 40, 0, 1)
    CALL CC_ADDMOB("DEFENDER", "冒険者くずれ１", 2000, 2000, 35)
    CALL CC_ADDMOB("DEFENDER", "ならず者１", 2000, 2000, 30)
    CALL CC_ADDMOB("DEFENDER", "ならず者２", 2000, 2000, 30)
    CALL CLOSE_COMBAT()
    ;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
    IF RESULT == 1
        CALL CC_ADD_COMBAT_EXP(6)
        CALL CC_APPLY_CHARA_HP()
        PRINTL
        PRINTFORMW 小部屋の敵を無力化することに成功した！
        CALL QST_SET_EVENT_FLAG(QST_NOWEV, 2)
    ELSE
        CALL CC_APPLY_CHARA_HP()
        CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%はならず者に返り討ちにされてしまった……", カラー_注意)
        PRINTFORMW 「ははっ、こいつぁは傑作だ。俺たちに勝てるとでも思ったのか？この間抜けめ」
        PRINTFORMW 「いまここでぶっ殺しちまってもいいが……まぁせっかくの新しい玩具だ、精々楽しませてもらうとしよう」
        PRINTFORMW 賊は敗北した%QST_LEADER_NAME()%を拘束すると、下卑た笑みを浮かべて弄びはじめた……
        CALL QST_QUEST_12_DEFEAT_EVENT
    ENDIF
ELSE
    PRINTFORMW 部屋の中は……がらんとしており、ひっくり返された椅子や乱雑にのけられた机から混乱の様子が伺える
    PRINTFORML 通風孔から送り込まれたガスの残り香か、微かな刺激臭が鼻孔をついた
    PRINTFORMW ……いずれにせよ脅威は排することができたようだ
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 2)
ENDIF

;-------------------------------------------------
;EV6 西の小部屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_6
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_12_EVENT_6

PRINTFORMW %QST_LEADER_NAME()%が入った部屋はことさら奇妙な空間であった
PRINTFORMW 天井に無数の配管が巡らされており、いくつかは壁際の装置類と接続されているのがみとめられる
PRINTFORMW これは何か気体を東側の部屋に流し込むための装置のようだ
PRINTL
PRINTFORMW 装置には実にわかりやすく粗野なドクロの模様が刻まれている。あとからならず者どもが刻んだのだろう
PRINTFORMW この装置が何か碌でもない作用を引き起こすものであることは間違いない
PRINTL
PRINTFORMW 配管に装置類、観察すればするほど部屋の異質さが際立ってくる
PRINTFORMW 今でこそ野盗たちの生活空間になっているものの、かつてはなにかを研究をするための施設だったのかもしれない
PRINTL
PRINTFORMW それが魔術なのか錬金術なのか、あるいはそのどちらでもない異質な技術なのか……調査の準備のない今は判断をつけられそうにない
PRINTL
PRINTFORMW さて、装置を稼働させれば東側の武器庫に何か影響を与えることができそうだ
;すでに東側の武器庫が排除されている
CALL QST_GET_EVENT_FLAG(5)
IF RESULT == 2
    PRINTFORMW ……が、武器庫の敵を殲滅した今、あえてこの不可解な装置に手を触れる理由もないだろう……
ELSE
    PRINTFORMW %QST_LEADER_NAME()%は古代の装置の稼働を試みても良いし、この怪しげな装置を見なかったことにしても良い
    PRINTL
    PRINTFORMW どうするにせよ、奥の大部屋に進む前に東の部屋、武器庫の小集団をまず掃除するのが賢明だ
    SIF QST_PARTYNUM > 1
        PRINTFORMW ……そう訴える%ANAME(QST_RANDCHARA())%の言葉に仲間は頷く
    PRINTL
    PRINTFORMW さもなければ大部屋での戦闘は一度に大勢の武装集団を相手取ることになるだろう……
ENDIF
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV7 奇妙な装置
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_7
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "奇妙な装置"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_12_EVENT_7

PRINTFORMW 装置は多数のツマミやレバーで構成されている
;すでに東側の武器庫が排除されている
CALL QST_GET_EVENT_FLAG(5)
IF RESULT == 2
    PRINTFORMW 今はあえてこの不可解な装置に手を触れる理由もないだろう……
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
    RETURN
ENDIF
PRINTFORMW さて……どうしたものだろうか？
CALL QST_ROLL_MULTI("知識,観察,身体,","解析を試みる(知識),よく観察してみる(観察),とりあえず叩いてみる(身体),", "50,100,150,", 7)
IF RESULT == 1
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW ……%ANAME(RESULT:1)%の見識によるとコモンウェルスの産業機械と類似点があり、操作できるかもしれないとのことだ
            PRINTFORMW そのような装置が古帝国以前に存在したという事実自体驚嘆すべきことだが、今は脅威の排除に集中する必要がある
            PRINTL
            PRINTFORMW %ANAME(RESULT:1)%が装置を操作すると機械類がゴトゴトと唸りを上げて稼働を始める
        CASE 1
            PRINTFORMW ……%ANAME(RESULT:1)%は装置と配管をじっくりと観察し、操作盤と各機械がどのように対応するのかアタリをつけていく
            PRINTFORMW そして直感を交えつつ装置を操作した
            PRINTL
            PRINTFORMW どうやら成功したようだ！%ANAME(RESULT:1)%が装置を操作すると機械類がゴトゴトと唸りを上げて稼働を始める
        CASE 2
            PRINTFORMW ……%ANAME(RESULT:1)%はお手上げと言わんばかりに肩をすくめると、装置を思いきりどついた！
            PRINTFORMW 少しの間をおいて機械類がゴトゴトと唸りを上げて稼働を始める。なんと成功してしまったようだ……
    ENDSELECT
    PRINTL
    PRINTFORMW 送風用と思われる回転翼が稼働を始め、配管を通して何かが隣室に送り込まれる
    PRINTFORMW やがてシュウシュウと噴出音が響き始めた
    PRINTL
    PRINTFORMW 「がぁぁぁぁ、目がぁ！目が痛ぇ！ああクソッ、何も見えねぇ！」
    PRINTFORMW 「馬鹿どもが……っ！機械を動かしやがったな！あれほど触るなって言い聞かせていたのによぉ！」
    PRINTL
    PRINTFORML 隣の部屋からはならず者たちが上げる絶叫が聞こえてくる
    PRINTFORMW 次いで乱雑に何かをひっくり返す音や破壊音、半狂乱にバタバタと部屋を飛び出していく足音が響く
    PRINTL
    PRINTFORMW ……どうやら労せず東の小部屋の脅威を排除することに成功したようだ！
    CALL QST_SET_EVENT_FLAG(5, 1)
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSEIF !RESULT
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW ……%ANAME(RESULT:1)%は装置を前にして首を捻っている。何もわかりそうにない……
        CASE 1
            PRINTFORMW ……%ANAME(RESULT:1)%は装置や配管を眺めてみたものの、結局時間だけが虚しく過ぎていった……
        CASE 2
            PRINTFORMW ……%ANAME(RESULT:1)%はやけくそ気味に装置をどついてみた。が、身体を襲う鈍い痛みの他は何も得られなかった……
    ENDSELECT
ELSEIF RESULT == -1
    PRINTFORMW ……得体の知れぬ装置をむやみに動かすべきではない、放っておこう
ENDIF
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV8 北の大部屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_8
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_8
#DIM EVENT_FLAG

PRINTFORMW %QST_LEADER_NAME()%が大部屋に押し入ると、室内にいた賊どもは一瞬虚ろを突かれたようだが、即座に応戦の構えを取った
PRINTL
PRINTFORMW 「チッ、ギルドの差し金か……！存外早かったな。おい野郎ども、ぐだぐだしてねぇでさっさと武器を取れ！」
PRINTFORMW 賊の頭領が檄をとばすや、斧や剣で武装した悪漢が%QST_LEADER_NAME()%の前に立ちはだかる
PRINTL
CALL CC_INIT()
CALL CC_ADDMOB("DEFENDER", "賊の頭領", 3000, 3000, 45, 0, 1)
CALL CC_ADDMOB("DEFENDER", "傭兵くずれ１", 2500, 2500, 40, 0, 1)
CALL CC_ADDMOB("DEFENDER", "ならず者１", 2000, 2000, 30)
CALL QST_GET_EVENT_FLAG(5)
;東の小部屋のならず者が排除されていない
IF !RESULT
    PRINTL
    PRINTFORMW さらに南側の入口から数名の賊が駆けつけて来た！
    PRINTFORMW 「お頭ぁ！なんの騒ぎで！？」
    PRINTFORMW 「……見りゃあわかるだろ、見りゃあよ。襲撃だ、襲撃。ボサッとしてねぇでとっとと加勢しやがれ！」
    PRINTL
    CALL CC_ADDMOB("DEFENDER", "傭兵くずれ２", 2500, 2500, 40, 0, 1)
    CALL CC_ADDMOB("DEFENDER", "冒険者くずれ１", 2000, 2000, 35)
    CALL CC_ADDMOB("DEFENDER", "ならず者２", 2000, 2000, 30)
ENDIF
PRINTFORMW ―――― 依頼を完遂するには連中をことごとく討滅する必要がある！
;戦闘イベント
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
NEXT
CALL CLOSE_COMBAT()
;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
IF RESULT == 1
    CALL CC_ADD_COMBAT_EXP(7)
    CALL CC_APPLY_CHARA_HP()
    PRINTL
    PRINTFORMW %ANAME(QST_RANDCHARA())%は賊の頭目の攻撃をかわし、すかさず痛烈な一閃を見舞う
    PRINTFORMW 痛恨の一撃を受けた頭領はもんどりを打ってその場に崩れ落ちた
    PRINTL
    PRINTFORMW 頭目を失った残党は互いに顔を見合わせると、蜘蛛の子を散らすように逃げ出していった……
    PRINTL
    PRINTL
    PRINTFORMW 依頼は完了だ！討伐したことを示す証拠となり得そうなものを回収した%QST_LEADER_NAME()%はこの場を立ち去った
    PRINTL
    ;クリア報酬
    CALL COLOR_PRINTW(@"――　ギルドから成功報酬として6,000ディナリが支払われた", カラー_注意)
    MONEY += 6000
    ;初回攻略時に求人書類を付与
    IF QST_SV_QUEST_FLAG:QST_有効クエストID == 0
        CALL COLOR_PRINTW(@"――　求人書類を1つ手に入れた！", カラー_注意)
        HO_ITEM_STOCK:ITEM_求人書類 += 1
    ENDIF
    CALL COLOR_PRINTW(@"――　名誉点を20点得た！", カラー_注意)
    QST_PRESTIGE_POINT += 20

    QST_COOLTIME:QST_有効クエストID += 4
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1

    ;リュミエール口上デイリーフラグ
    IF !K1_DFLAG_メイドの嗜み && LIST_INDEXOF(_LIST_QST_PARTY, TOSTR(NAME_TO_CHARA("リュミエール"))) != -1
        K1_DFLAG_メイドの嗜み = 1
    ENDIF
    TFLAG:クエスト_完了フラグ = 1
    CALL TRAIN_EXIT()
ELSE
    CALL CC_APPLY_CHARA_HP()
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%はならず者に返り討ちにされてしまった……", カラー_注意)
    PRINTFORMW 「ギルドの質も落ちたもんだな。てんで歯ごたえがねぇ」
    PRINTFORMW 「さっさとぶっ殺しちまってもいいが……部下どもの"はけ口"には丁度いいだろう」
    PRINTFORMW 賊は敗北した%QST_LEADER_NAME()%を拘束すると、下卑た笑みを浮かべて弄びはじめた……
    CALL QST_QUEST_12_DEFEAT_EVENT
ENDIF
;-------------------------------------------------
;EV100 宝箱(技巧の宝玉)
;-------------------------------------------------
;イベント登録
@QST_ADD_EVENT_QUEST_12_EVENT_100
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"木箱"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_12_EVENT_100
#DIMS DYNAMIC _ITEM_LIST
#DIM  DYNAMIC CHEST_TYPE

_ITEM_LIST '= @"{ITEM_技巧の宝玉},1,"
CALL QST_CEV_CHEST(_ITEM_LIST)
IF RESULT
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;-------------------------------------------------
;EV101 宝箱(ポーション)
;-------------------------------------------------
;イベント登録
@QST_ADD_EVENT_QUEST_12_EVENT_101
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"施錠された木箱"
;取得済みの場合は非アクティブに
CALL QST_GET_EVENT_FLAG(QST_NOWEV)
DEBUGPRINTFORML RESULT:{RESULT}
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_12_EVENT_101
#DIMS DYNAMIC _ITEM_LIST
_ITEM_LIST '= @"{ITEM_ポーション},2,"
CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST)
IF RESULT
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;-------------------------------------------------
;EV102 宝箱(衣服)
;-------------------------------------------------
;イベント登録
@QST_ADD_EVENT_QUEST_12_EVENT_102
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"木箱"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

;イベント実行内容
@QST_QUEST_12_EVENT_102
PRINTFORMW 木箱の中から略奪品と思われる衣類を見つけた
CALL CHANGE_CLOTH_STOCK(衣装部位_上下服, 15, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_上下服, 10, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_上下下着, 7, 1)
CALL PRINT_NOTICE("衣服:ぼろぼろの布切れを手に入れた")
CALL PRINT_NOTICE("衣服:ディアンドルを手に入れた")
CALL PRINT_NOTICE("衣服:ネグリジェを手に入れた")
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
CALL QST_DEACTIVATE_EVENT(QST_NOWEV)

;-------------------------------------------------
;EV10
;敗北イベント後の共通エンディング（クエスト失敗）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_10
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_10
#DIM CONST 派生イベントID = 11
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC HIGHEST_SEDUCTION_LV

PRINTL
PRINTFORMW 「あ？これで終わりだとでも思ったのか？」
PRINTFORMW 賊の頭領は激しい輪姦を受け息絶え絶えになる%QST_LEADER_NAME()%を引きずり上げると、首枷をはめ、身体の自由を奪う……
PRINTL
PRINTFORMW 「俺たちもなぁ……こんな狭苦しいところに野郎だけで押し込められて、溜まってんだよ」
PRINTFORMW 「これからは肉便器として散々使ってやるよ。感謝しな、命までは取らねぇおいてやる。今はな」
PRINTL
PRINTFORMW 「もっとも……部下どもは荒っぽくてなぁ、もしかしたら輪姦されてるうちに死んじまうかもしれん」
PRINTFORMW 「まぁそんときは自分の力不足を呪ってくれや、はははっ」
PRINTL
PRINTFORMW 「さーて……んじゃ、始めるか。お前ら、戦いの褒美だ。こいつらは好きにしていいぞ！」
PRINTL
PRINTFORMW 「さっすがお頭ぁ！話がわかりやすぜ」
PRINTFORMW 色めき立った賊たちが疲弊する%QST_LEADER_NAME()%の肢体に群がる……

;パーティの女性メンバーの誘蛾の色香LVの最高値を算出
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF IS_FEMALE(CHARA_ID)
        CALL SKILL_GET_LEVEL(CHARA_ID, スキル_大分類_性技, "SEDUCTION")
        SIF RESULT > HIGHEST_SEDUCTION_LV
            HIGHEST_SEDUCTION_LV = RESULT
    ENDIF
NEXT

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
;パーティを追加
FOR LOCAL, 0, QST_PARTYNUM
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):凌辱被害者フラグ = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):調教参加フラグ = 1
NEXT

;竿役生成
FOR LOCAL, 0, 4
    CALL QST_QUEST_12_CREATE_RAPER(0)
    CHARA_ID = RESULT
    CFLAG:CHARA_ID:調教参加フラグ = 1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
    IF HIGHEST_SEDUCTION_LV > 0
        MAXBASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力
        MAXBASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力
        ABL:CHARA_ID:倒錯度 =  ABL:CHARA_ID:倒錯度 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性技 = ABL:CHARA_ID:性技 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:欲望 = ABL:CHARA_ID:欲望 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性交 = ABL:CHARA_ID:性交 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:射精 = ABL:CHARA_ID:射精 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:Ｃ感 = ((ABL:CHARA_ID:射精 + ABL:CHARA_ID:欲望))
        IF RAND(2)
            TALENT:CHARA_ID:Ｃ敏感 = 1
            TALENT:CHARA_ID:絶倫 = 1
        ELSE
        ENDIF
    ENDIF
NEXT
;適当に被害者を追加
FOR LOCAL, 0, 2
    CALL QST_QUEST_12_CREATE_TRAIN_VICTIMS
NEXT
FOR LOCAL, 0, CHARANUM
    IF CFLAG:LOCAL:慰安参加者
        ;立ちバック系のみ許可
        VARSET COM_FORBIDDEN:LOCAL:0, 1
        COM_FORBIDDEN:LOCAL:T_N("服を破る") = 0
        COM_FORBIDDEN:LOCAL:T_N("二本挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("立ちバック") = 0
        COM_FORBIDDEN:LOCAL:T_N("順番に挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("代わる代わる挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("Ａ立ちバック") = 0
        COM_FORBIDDEN:LOCAL:T_N("順番にＡ挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("Ａ二本挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("イラマチオ") = 0
    ENDIF
NEXT

FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
IF HIGHEST_SEDUCTION_LV > 0
    TFLAG:調教ターン数 = 20 +  LIMIT(HIGHEST_SEDUCTION_LV ,1 , 3) * 4
ELSE
    TFLAG:調教ターン数 = 20
ENDIF

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
RETURN

;-------------------------------------------------
;EV11
;敗北イベント後の共通エンディング（クエスト失敗）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_11
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_11
#DIM CONST EVENT_ID = 5
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
PRINTFORMW ……一体どれほど犯されたのだろうか？
PRINTFORMW %QST_LEADER_NAME()%はならず者どもに散々に辱められてしまった……
PRINTFORML 虜囚として散々な辱めを受けた後、幸運にもスキを突いて脱出する機会を得た%QST_LEADER_NAME()%は
PRINTFORMW 体液にまみれ、ぐったりと重い身体からなけなしの力を振り絞り、脱走を図った
PRINTL
PRINTFORMW なんとか近隣の町まで戻ることに成功した……
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
@QST_QUEST_12_DEFEAT_EVENT
#DIM CONST 派生イベントID = 10
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC HIGHEST_SEDUCTION_LV

FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 12

;パーティの女性メンバーの誘蛾の色香LVの最高値を算出
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF IS_FEMALE(CHARA_ID)
        CALL SKILL_GET_LEVEL(CHARA_ID, スキル_大分類_性技, "SEDUCTION")
        SIF RESULT > HIGHEST_SEDUCTION_LV
            HIGHEST_SEDUCTION_LV = RESULT
    ENDIF
NEXT

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):凌辱被害者フラグ = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):調教参加フラグ = 1
NEXT
;適当に被害者を追加
FOR LOCAL, 0, 2
    CALL QST_QUEST_12_CREATE_TRAIN_VICTIMS
NEXT
;竿役生成
FOR LOCAL:0, 0, 3
    CALL QST_QUEST_12_CREATE_RAPER(0)
    CHARA_ID = RESULT
    CFLAG:CHARA_ID:調教参加フラグ = 1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
    IF HIGHEST_SEDUCTION_LV > 0
        MAXBASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力
        MAXBASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力
        ABL:CHARA_ID:倒錯度 =  ABL:CHARA_ID:倒錯度 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性技 = ABL:CHARA_ID:性技 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:欲望 = ABL:CHARA_ID:欲望 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性交 = ABL:CHARA_ID:性交 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:射精 = ABL:CHARA_ID:射精 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:Ｃ感 = ((ABL:CHARA_ID:射精 + ABL:CHARA_ID:欲望))
        IF RAND(2)
            TALENT:CHARA_ID:Ｃ敏感 = 1
            TALENT:CHARA_ID:絶倫 = 1
        ELSE
        ENDIF
    ENDIF
NEXT

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
RETURN

;-------------------------------------------------
;慰安モブの定義
;-------------------------------------------------
@QST_QUEST_12_CREATE_TRAIN_VICTIMS()
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %SPLIT_RAND("冒険者/村娘/行商人/性奴隷",1)%

CALL CREATE_RANDOM_CHARA(0, -1, 1)
FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
CALLNAME:対象 = %名前%
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = 1
ABL:対象:性知識 =  3
ABL:対象:Ｃ感 = 28
ABL:対象:奉仕 = 1
ABL:対象:性技 = RAND(10, 14)
ABL:対象:欲望 = RAND(3, 7)
ABL:対象:性交 = RAND(13, 15)
ABL:対象:マゾ = RAND(0, 3)
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = RAND(0, 300)
CFLAG:対象:調教参加フラグ  = 1
CFLAG:対象:主導除外  = 1
CFLAG:対象:慰安モブ = 1
CFLAG:対象:凌辱被害者フラグ = 1

;-------------------------------------------------
;慰安モブの定義
;-------------------------------------------------
@QST_QUEST_12_CREATE_RAPER(SEX)
#DIM SEX
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %SPLIT_RAND("ならず者/悪漢/野盗/傭兵くずれ/盗賊/冒険者くずれ",1)%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)
FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
CALLNAME:対象 = %名前%
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
ABL:対象:戦闘 =  RAND(35, 50)
ABL:対象:性知識 = 3
ABL:対象:奉仕 = 1
ABL:対象:性技 = RAND(0, 4)
ABL:対象:欲望 = RAND(3, 7)
ABL:対象:性交 = RAND(3, 5)
ABL:対象:射精 = RAND(3, 5)
ABL:対象:サド = RAND(4, 8)
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = RAND(0, 300)
CFLAG:対象:調教参加フラグ = 1
CFLAG:対象:慰安モブ = 1

COM_TENDENCY:対象:GETNUM(TRAINNAME, "服を破る") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "代わる代わる挿入") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "代わる代わるＡ挿入") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "順番に挿入") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "順番にＡ挿入") = 3

LOCALS = %S_R("male01/male02/male03/shota01/shota02")%
CSTR:対象:顔グラファイル名 '= @"%LOCALS%.webp"
CSTR:対象:キャラ顔グラ '= @"%LOCALS%.webp"
CFLAG:対象:ポートレイト画像タイプ = 2
CFLAG:対象:汎用口上パターン = 2002
CFLAG:対象:口上カラーコード = カラー_選択不可

RETURN 対象


code:
MAP_NAME = "盗賊の拠点"
MAP_IMAGE = "dg00.webp"
PATH_COLOR = "0xff00"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0x2c2c2c"


code:
<DocumentElement>
  <GMAPDATA>
    <id>638135971342404270</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>エントランス</NODE_NAME>
    <POS_X>160</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971426815879</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>0,2,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971448656173</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>140</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971479074879</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>280</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>1,5,6,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971642223563</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>140</POS_Y>
    <PATH_LIST>6,7,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971679283709</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>180</POS_Y>
    <PATH_LIST>5,7,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971722553106</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>大部屋</NODE_NAME>
    <POS_X>480</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>6,5,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_13:辺境の教会
;-------------------------------------------------
@QST_TITLE_QUEST_13
RESULTS '= "辺境の教会"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_13
RESULTS:0 '= @"教会は来るものを拒まない。もし%ANAME(MASTER)%が望むのならば足を運んでもよいだろう"
RESULTS:1 '= "望むなら祈りを捧げたり、後ろめたい罪があるあらば告白しても良い"
RESULTS:4 '= "古帝国が正教を定めて以来、中央大陸各地に教会が建築され土着の信仰に取って代わった"
RESULTS:5 '= "帝国亡き後も正教は大陸の人々に広く信仰され、教会は地域の信仰の拠点となっている"
RESULTS:6 '= "ただし帝国の崩壊によって中央による指導が行き届かなくなると土着の宗教と教義が入り混じったり"
RESULTS:7 '= "異端化・カルト化し変質するケースも少なくはないようだ"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_13
RESULTS:0 '= @"難易度：@C:{カラー_注意}@命の危険はおそらくない@/C@"
RESULTS:1 '= "報酬：-"
RESULTS:2 '= "推奨パーティ人数:1人以上"
RESULTS:3 '= ""
RESULTS:4 '= @"@C:{カラー_選択不可}@「なんだお前さんそんなに信心深かったのか？」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「祈ってる手が空いているなら剣でも握ってればいいだろうに」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_13(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_0_ゴブリンの洞窟の攻略が必要
SIF !QST_SV_QUEST_FLAG:0
    RETURN 0
SIF QST_COOLTIME:QUEST_ID
    RETURN 0
RETURN 1

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_13(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_13_辺境の教会"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "辺境の教会"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,2,3,4,]"
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,50,51,59,]"

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_13
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_13

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_13_ABORT
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_13
; PRINTL
; PRINTFORMW ……………………
; PRINTFORMW ………………
; PRINTFORMW …………
; PRINTL

; PRINTFORML %QST_LEADER_NAME()%は領地の外れにある教会に足を運んだ
; PRINTFORMW 最低限の礼拝堂を備えた小さな教会で、人の気配はほぼ無いに等しい
; PRINTL
; PRINTFORMW 村人の話によると今は若い姉妹のシスターが二人で切り盛りしているとのことだ

;ゲストユニークキャラ、ロザリィを追加
LOCAL = NAME_TO_CHARA("シスター・ロザリィ")
IF LOCAL == -1
    CALL QST_QUEST_13_CREATE_UNIQUE_MOB1()
    LOCAL = RESULT
ENDIF
CALL QST_SET_FLAGMAP("NPC_ID", TOSTR(LOCAL))
;クエスト用フラグ管理変数にCHARA_IDを登録

;ゲストユニークキャラ、シャーリィを追加
LOCAL = NAME_TO_CHARA("シスター・シャーリィ")
IF LOCAL == -1
    CALL QST_QUEST_13_CREATE_UNIQUE_MOB2()
    LOCAL = RESULT
ENDIF
;クエスト用フラグ管理変数にCHARA_IDを登録
CALL QST_SET_FLAGMAP("NPC_ID2", TOSTR(LOCAL))

FLAG:クエスト_中断を許可 = 1

;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_13_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 1
        CALL HO_PRINT_LOG("ホールは静寂に包まれている。人の気配は無い", 1)
    CASE 2
        CALL HO_PRINT_LOG("部屋の一角に厳しい装飾の施された懺悔室がぽつんと置かれている", 1)
    CASE 4
        CALL HO_PRINT_LOG("薄暗い部屋を燭台の頼りない明かりが照らしている", 1)
    CASE 5
        CALL HO_PRINT_LOG("薄暗く、静寂に包まれた礼拝堂で僅かな信徒が祈りを捧げている", 1)
ENDSELECT

;-------------------------------------------------
;EV2
;ロザリィに告解
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_2
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "懺悔室で告解を始める"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_2
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC NPC_ID

; IF RAND:2
;     NPC_ID = NAME_TO_CHARA("シスター・ロザリィ")
; ELSE
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
; ENDIF
CALL QST_SET_FLAGMAP("NPC_ID", TOSTR(NPC_ID))
CALL QST_GET_EVENT_FLAG(QST_NOWEV)
IF !RESULT
    PRINTFORML 懺悔室がある。望むならシスター・%ANAME(NPC_ID)%に悩み事や罪を告白しても良い
    CALL QST_SELECT_CHARA(, "さて誰に告解させようか……")
    CHARA_ID = RESULT
    IF CHARA_ID == -1
        CALL KJ_PRINT(NPC_ID, "神はいつでも貴方の言葉に耳を傾けています。もし自身の歩む道に疑問が生じたなら、いつでもいらしてください", 1)
    ELSE
        CALL QST_SET_FLAGMAP("CHARA_ID", TOSTR(CHARA_ID))
        CALL QST_ACTIVATE_EVENT(3)
        CALL QST_ACTIVATE_EVENT(4)
        CALL QST_SET_EVENT_SLOT_NAME(, "懺悔室を出る")
        CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
        ;移動を完全に制限
        FLAG:移動制限フラグ = 2
        CALL KJ_PRINT(NPC_ID, "貴方の迷いをお聞きいたしましょう。さぁ、お入りになって", 1)
    ENDIF
ELSE
    CALL QST_DEACTIVATE_EVENT(3)
    CALL QST_DEACTIVATE_EVENT(4)
    CALL QST_SET_EVENT_SLOT_NAME(, "懺悔室で告解を始める")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    ;移動制限を解除
    FLAG:移動制限フラグ = 0
    CALL KJ_PRINT(NPC_ID, @"%ANAME(CHARA_ID)%、神はいつでも貴方の言葉に耳を傾けていますよ。もし自身の歩む道に疑問が生じたなら、いつでもいらしてください", 1)
ENDIF

;-------------------------------------------------
;EV3
;ロザリィに告解
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_3
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "告解する"

CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_3
#DIM NPC_ID
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))
CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MTAR(CHARA_ID)
;CALL QST_EXEC_DUMMYCOM(T_N("フェラチオ"))

;MASTERが対象ならソース処理は会話と同じものを使う
IF CHARA_ID == MASTER
    CALLFORM COM{T_N("会話")}
ELSE

ENDIF
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は懺悔室で%ANAME(NPC_ID)%に自身の悩みを打ち明けた
;擬似コマンド実行
CALL QST_EXEC_DUMMYCOM()

;-------------------------------------------------
;EV4
;ロザリィに告解フェラ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_4
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = @"@C:{カラー_ピンク}@❤告解する❤@/C@"

CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_4
#DIM EVENT_FLAG
#DIM NPC_ID
#DIM NPC_ID2
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
NPC_ID2 = TOINT(QST_GET_FLAGMAP("NPC_ID2"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))

CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
CFLAG:NPC_ID:スケベモブ = 1
IF NPC_ID2
    CFLAG:NPC_ID2:調教参加フラグ = 1
    CFLAG:NPC_ID2:強制友好化 = 1
    CFLAG:NPC_ID2:スケベモブ = 1
ENDIF
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MPLY(NPC_ID2)
CALL ADD_MTAR(CHARA_ID)

;ソース処理は会話と同じものを使う
CALLFORM COM{T_N("会話")}
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は懺悔室で%ANAME(NPC_ID)%に自身の"悩み"を処理させた……
IF HAS_PENIS(CHARA_ID)
    ;擬似コマンド実行
    CALL QST_EXEC_DUMMYCOM(T_N("フェラチオ"))
ELSE
    CALL QST_EXEC_DUMMYCOM(T_N("クンニ"))
ENDIF

;ダミーコマンド実行後に呼び出されるイベント
@QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT
#DIM NPC_ID
#DIM NPC_ID2
#DIM 射精量
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
NPC_ID2 = TOINT(QST_GET_FLAGMAP("NPC_ID2"))

射精量  = HO_CHECK_CUM_AMOUNT(NPC_ID, 0)
射精量  += HO_CHECK_CUM_AMOUNT(NPC_ID2, 0)
IF 射精量 >= 200
    CALL QST_SET_ROOM_EFFECT("むせ返るような精臭", "0")
ENDIF
CFLAG:NPC_ID:調教参加フラグ = 0
CFLAG:NPC_ID:強制友好化 = 0
CFLAG:NPC_ID:スケベモブ = 0
IF NPC_ID2
    CFLAG:NPC_ID2:調教参加フラグ = 0
    CFLAG:NPC_ID2:強制友好化 = 0
    CFLAG:NPC_ID2:スケベモブ = 0
ENDIF
FLAG:主導権所有者 = -1
CALL INIT_EQUIP()
CALL CLEAR_MPLY
CALL CLEAR_MTAR

IF GROUPMATCH(SELECTCOM, T_N("フェラチオ"), T_N("クンニ"))
    {
        LOCALS '=
            @"%S_R("密室の中で/厳かで窮屈な空間の中で/一枚の薄い壁の向こうで/壁を隔てた向こうで")%"
            +@"%S_R("ぴちゃぴちゃと/にちゃにちゃと/くちゅくちゅと")%肉を啜る、"
            +@"%S_R("淫猥な水音/淫らな音/色に満ちた音")%"
            +@"が響いている"
    }
    CALL COLOR_PRINTL(@"%LOCALS%", カラー_ピンク)
ENDIF
; #endregion

;-------------------------------------------------
;礼拝堂
;-------------------------------------------------
;-------------------------------------------------
;ロザリィと話す
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_50
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ロザリィと話す"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_50
#DIM NPC_ID
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))
CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MTAR(CHARA_ID)

;MASTERが対象ならソース処理は会話と同じものを使う
IF CHARA_ID == MASTER
    CALLFORM COM{T_N("会話")}
ELSE
ENDIF
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は礼拝堂で%ANAME(NPC_ID)%と談笑した
;擬似コマンド実行
CALL QST_EXEC_DUMMYCOM()

;-------------------------------------------------
;シャーリィと話す
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_51
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "シャーリィと話す"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_51
#DIM NPC_ID
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID2"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))
CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MTAR(CHARA_ID)

;MASTERが対象ならソース処理は会話と同じものを使う
IF CHARA_ID == MASTER
    CALLFORM COM{T_N("会話")}
ELSE
ENDIF
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は礼拝堂で%ANAME(NPC_ID)%と談笑した
;擬似コマンド実行
CALL QST_EXEC_DUMMYCOM()

@QST_ADD_EVENT_QUEST_13_EVENT_59
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "信徒席"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_59
#DIM EVENT_FLAG
#DIM CHARA_ID

;擬似設備の使用
;MODULE_ID:5(机)を擬似的に使用している状態に
{
    CALL QST_CEV_USE_DUMMY_MODULE(
        @"[MODULE_ID:5]"
        +@"[PAUSE_BGM:1]"
        +@"[ALLOW_COM:1]"
        +@"[MODULE_NAME:信徒席]"
        +@"[USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は信徒席に座った……]"
        +@"[MODULE_NAME_AT_USING:席を立つ]"
        +@"[STOP_USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は席を立った]"
        +@"[REDRESS_AT_STOP:1]"
        )
}

;-------------------------------------------------
;クエスト用ユニークNPC
;-------------------------------------------------
@QST_QUEST_13_CREATE_UNIQUE_MOB1()
#DIM SEX = 1
#DIMS MOB_NAME = "ロザリィ"
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)

FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
;ユニークキャラとして一度だけ生成するのでCSTR:99を登録
CSTR:対象:99 '= "シスター・ロザリィ"
CALLNAME:対象 = %名前%
BASE:対象:体力 = 2500
BASE:対象:気力 = 2500
MAXBASE:対象:体力 = 2500
MAXBASE:対象:気力 = 2500
TALENT:対象:処女 = 1
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
TALENT:対象:絶倫 = 1

CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
ABL:対象:戦闘 =  45
ABL:対象:魔術 =  0
ABL:対象:性知識 = 5
ABL:対象:奉仕 = 6
ABL:対象:精愛 = 7
ABL:対象:性技 = 6
ABL:対象:欲望 = 7
ABL:対象:性交 = 4
ABL:対象:サド = 4
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = 500

CSTR:対象:120 '= "CC1_後髪_5"
CSTR:対象:121 '= "CC1_身体_0"
CSTR:対象:122 '= "CC1_顔_0"
CSTR:対象:123 '= "CC1_瞳_2"
CSTR:対象:124 '= "CC1_目_2"
CSTR:対象:128 '= "CC1_服_ローブ_0"
CSTR:対象:129 '= "CC1_前髪_4"
CSTR:対象:150 '= "384:344:288"
CSTR:対象:151 '= "白"
CSTR:対象:152 '= "白"
CSTR:対象:153 '= "510:100:204"
CSTR:対象:158 '= "256:248:348"
CSTR:対象:159 '= "384:344:288"

CFLAG:対象:汎用口上パターン = 2010
CFLAG:対象:口上カラーコード = カラー_クリーム

TALENT:対象:バストサイズ = 2
TALENT:対象:ヒップサイズ = 1
TALENT:対象:体格 = 0

CALL SET_BODYSIZE(対象)

RETURN 対象

@QST_QUEST_13_CREATE_UNIQUE_MOB2()
#DIM SEX = 1
#DIMS MOB_NAME = "シャーリィ"
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)

FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
;ユニークキャラとして一度だけ生成するのでCSTR:99を登録
CSTR:対象:99 '= "シスター・シャーリィ"
CALLNAME:対象 = %名前%
BASE:対象:体力 = 2500
BASE:対象:気力 = 2500
MAXBASE:対象:体力 = 2500
MAXBASE:対象:気力 = 2500
TALENT:対象:処女 = 1
TALENT:対象:キス未経験 = 1
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 1
TALENT:対象:絶倫 = 1
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
ABL:対象:戦闘 =  45
ABL:対象:魔術 =  0
ABL:対象:性知識 = 5
ABL:対象:奉仕 = 6
ABL:対象:精愛 = 0
ABL:対象:性技 = 0
ABL:対象:欲望 = 0
ABL:対象:性交 = 0
ABL:対象:サド = 0
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = 250

CSTR:対象:120 '= "CC1_後髪_11"
CSTR:対象:121 '= "CC1_身体_0"
CSTR:対象:122 '= "CC1_顔_0"
CSTR:対象:123 '= "CC1_瞳_2"
CSTR:対象:124 '= "CC1_目_2"
CSTR:対象:128 '= "CC1_服_ローブ_2"
CSTR:対象:129 '= "CC1_前髪_22"
CSTR:対象:130 '= "CC1_眉_0"
CSTR:対象:132 '= "CC1_帽子_ヴェール_2"
CSTR:対象:150 '= "黄"
CSTR:対象:151 '= "白"
CSTR:対象:152 '= "白"
CSTR:対象:153 '= "赤"
CSTR:対象:158 '= "白"
CSTR:対象:159 '= "黄"

CFLAG:対象:汎用口上パターン = 2010
CFLAG:対象:口上カラーコード = カラー_クリーム

TALENT:対象:バストサイズ = -1
TALENT:対象:ヒップサイズ = 0
TALENT:対象:体格 = -1

CALL SET_BODYSIZE(対象)

RETURN 対象


code:
MAP_NAME = "辺境の教会"
MAP_IMAGE = "dg08.webp"
PATH_COLOR = "0xff00"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0x4834FF"

path: ERB/QUEST/クエストデータ/QUEST_13_辺境の教会/辺境の教会.txt


code:
<DocumentElement>
  <GMAPDATA>
    <id>638137465039467278</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>入口</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465088867759</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>エントランスホール</NODE_NAME>
    <POS_X>260</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>0,4,2,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465137187875</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>礼拝堂-右翼</NODE_NAME>
    <POS_X>360</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST>1,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465168817767</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>懺悔室</NODE_NAME>
    <POS_X>440</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST>2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465199298193</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>礼拝堂-左翼</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>120</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465241837690</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>祭壇</NODE_NAME>
    <POS_X>440</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_13_辺境の教会/辺境の教会.xml


code:
;露店の商品在庫(商品在庫を管理・セーブする必要がない場合は、ERHへの定義は控えた方が良い)
#DIMS SAVEDATA QUEST_14_DIC_SHOP_STOCK_0, 10
#DIMS SAVEDATA QUEST_14_DIC_SHOP_STOCK_1, 10
#DIMS SAVEDATA QUEST_14_DIC_SHOP_STOCK_2, 10
;ダートトラックミニゲーム用の変数
#DIMS QUEST_14_C_Name